
Change 1 – IPCAN_RAN_Type, IMS_SecurityScheme_Type
	  type enumerated IPCAN_RAN_Type {              /* radio access network technology used by the IPCAN PTC */

    GERAN,

    UTRAN_FDD,

    UTRAN_TDD,

    EUTRA_FDD,

    EUTRA_TDD,

    WLAN,                                       /* @sic R5-163201 sic@ */


    NR,                                         /* @sic R5-188107 sic@ */

    MCX_IP                                      /* @sic R5-203077: MCX IP test model sic@ */

  };

	  type enumerated IMS_SecurityScheme_Type {

    GIBA,               /* 34.229-2 Table A.6a Item 1 */

    IMS_Security       /* 34.229-2 Table A.6a Item 2 */


  };


Change 2 – Removal of f_IMS_PTC_Security_InitFBBA
	  function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,

                                       B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_StartSecurity := true) runs on IMS_PTC

  { 

    var Common_AuthenticationParams_Type v_AuthenticationParams;

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var template (omit) PortNumber_Type v_UnprotectedPort_us := f_SIP_ContactUrl_GetPort(v_ContactUrl);           // may be omit in which case 5060 is chosen by the IP PTC


    if (ispresent(p_RegisterReq.msgHeader.securityClient)) {                              /* @sic R5s150873: false for GIBA sic@ */

      f_IMS_Register_SecurityInit_ImsSecurity(p_RegisterReq, p_Rand, p_StartSecurity);    /* @sic R5s190727 change 1.5: MCX split sic@ */

    }

    else { /* GIBA */





      f_IMS_PortsAndSecurityConfig(v_UnprotectedPort_us, cs_IMS_RegistrationInfo(v_NW_Address, v_UE_Address));

    }

  }


Change 3 – f_IMS_PTC_Init

	  function f_IMS_PTC_Init(PDN_Index_Type p_PdnIndex := PDN_1,

                          IMS_PTC_Handling_Type p_Handling := testcaseSpecificHandling,

                          template (omit) IMS_SecurityScheme_Type p_SecurityScheme := omit,

                          template (omit) IPCAN_RAN_Type p_RanType := omit,


                          IMS_TestSpec_Type p_TestSpec := TS_34229,

                          NameValueList_Type p_TestsuiteParameterList := {})  runs on IMS_PTC

  {

    var charstring v_HomeDomainName;

    var charstring v_PrivateUserId;

    var CharStringList_Type v_IMPUList;

    var IMS_SecurityScheme_Type v_SecurityScheme;

    var charstring v_Password_FBBA;



    var IntegerList_Type v_SPI_ValidValueList_PDN1 := fl_IMS_PTC_SPI_ValidValueList_Init(px_IMS_SPI_ValidValueList_PDN1);  /* @sic R5-172048 sic@ */

    var IntegerList_Type v_SPI_ValidValueList_PDN2 := fl_IMS_PTC_SPI_ValidValueList_Init(px_IMS_SPI_ValidValueList_PDN2);  /* @sic R5-172048 sic@ */

    var template (omit) integer v_UE_Release := omit;

    var boolean v_UEwithISIM := px_UEwithISIM;

    f_IMS_PTC_TestsuiteInfo_Init(p_TestSpec, p_TestsuiteParameterList);

    v_IsMCX := f_IMS_PTC_TestsuiteInfo_IsMCX();

    if (v_IsMCX) {   /* @sic R5-196983 sic@ */

      v_UEwithISIM := not match(f_IMS_PTC_TestsuiteInfo_GetParameter("MCX_SIP_RegistrationWithTemporaryIdentities"), "true");

      if (v_UEwithISIM) {

        v_HomeDomainName := f_IMS_PTC_TestsuiteInfo_GetParameter("MCX_SIP_HomeDomain_A");

        v_PrivateUserId  := f_IMS_PTC_TestsuiteInfo_GetParameter("MCX_SIP_PrivateUserId_A");

      } else {

        v_HomeDomainName := f_DomainName_IMS_DerivedFromIMSI(px_IMSI_Def);

        v_PrivateUserId := f_IMS_PrivateUserIdentityDerivedFromIMSI(px_IMSI_Def);

      }

    }

    else if (px_UEwithISIM ) {  /* @sic R5-133628: check of temporary user id sic@ */

      v_HomeDomainName := px_IMS_HomeDomainName;

      v_PrivateUserId := px_IMS_Private_UserId;

    }

    else {

      v_HomeDomainName := f_DomainName_IMS_DerivedFromIMSI(px_IMSI_Def);

      v_PrivateUserId := f_IMS_PrivateUserIdentityDerivedFromIMSI(px_IMSI_Def);

    }

    if (ispresent(p_SecurityScheme)) {

      v_SecurityScheme := valueof(p_SecurityScheme);

    } else if (pc_IMS_GIBA_Sec and not pc_IMS_Sec) {

      v_SecurityScheme := GIBA;

    } else if (pc_IMS_Sec) {

      v_SecurityScheme := IMS_Security;

    } else {

      FatalError(__FILE__, __LINE__, "Invalid security setting (neither pc_IMS_GIBA_Sec nor pc_IMS_Sec is set)");

    }

    

if (v_IsMCX) {   /* @sic R5-196983 sic@ */

      v_IMPUList := { f_IMS_PTC_TestsuiteInfo_GetParameter("MCX_SIP_PublicUserId_A_1") };

    } else {

      v_IMPUList := {px_IMS_PublicUserIdentity1, px_IMS_PublicUserIdentity2, px_IMS_PublicUserIdentity3};

    }

    vc_IMS_Global.PdnIndex := p_PdnIndex;

    vc_IMS_Global.Ctrl.TestBody := false;

    vc_IMS_Global.Ctrl.BlockUlDataInDefault := false;       /* @sic R5s140780 - MCC160 Implementation sic@ */

    vc_IMS_Global.Ctrl.IgnoreReset := true;                 /* @sic R5s150031: IMS is already reset after initialisation sic@ */

    vc_IMS_Global.Ctrl.Default := null;

    vc_IMS_Global.Security := valueof(cs_IMS_SecurityInfo_Init);









    vc_IMS_Global.ImsInfo  := valueof(cs_IMS_ImsInfo_Init(p_RanType, v_IMPUList, v_UE_Release, v_HomeDomainName, v_PrivateUserId, v_SecurityScheme, v_SPI_ValidValueList_PDN1, v_SPI_ValidValueList_PDN2, v_UEwithISIM));

    f_IMS_PTC_ActivateDefault(p_Handling);

  }


Change 4 – f_IMS_MOCallSetup_ReceiveINVITE_Optional
	  function f_IMS_MOCallSetup_ReceiveINVITE_Optional(template (omit) float p_WaitDuration := omit,

                                                    template SipUrl p_OtherCalleeUri := omit,

                                                    template SDP_Message p_SDP_Message := omit,

                                                    IMS_SDP_AudioCodecList_Type p_AudioCodecList,

                                                    boolean p_PreconditionSupported := true,

                                                    boolean p_OIR := false,

                                                    boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type

  {

    var OptionTag_List v_AdditionalOptionTagsForSupported := f_IMS_AdditionalOptionTagsForSupported(p_PreconditionSupported);  /* @sic R5s180197 sic@ */

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (omit) IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := omit;

    var template (present) SDP_Message v_SDP_MessageTemplate;

    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A4;

    var template (present) SipUrl v_ExpectedCalleeUri := f_IMS_INVITE_CalleeUriRX(p_OtherCalleeUri, v_Context);

    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, v_AdditionalOptionTagsForSupported, v_Context, -, p_GeolocationInfoAvailable);  // @sic R5s190882 sic@

    var IMS_DATA_REQ v_IMS_DATA_REQ_Value;

    var SDP_Message v_SDP_Message;

    var INVITE_Request v_InviteRequest;


    var boolean v_CheckForbiddenCodecParameters;                /* @sic R5-180231 sic@ */

    if (ispresent(p_SDP_Message)) {                             /* specific SDP message content @sic R5s180197 sic@ */

      v_SDP_MessageTemplate := p_SDP_Message;

      v_CheckForbiddenCodecParameters := false;                 /* no automatic check for forbidden parameters in case of specific (i.e. non C.21(abcd)) SDP messages */

    }

    else {                                                      /* SDP message content according to C.21(abcd) */

      v_SDP_MessageTemplate := f_IMS_BuildSDP_InitialOfferAudio_RX(p_PreconditionSupported, p_AudioCodecList); /* @sic R5s150704 - additional changes sic@ @sic R5-172931 sic@ */




        v_CheckForbiddenCodecParameters := true;                /* automatic check for forbidden parameters for Cat-M1 (C.21d), EVS (C.44), WLAN (C.21a) and normal cases (C.21);

                                                                   @sic R5-180231: grace period, R5-192309: grace period for normal cases expired sic@ */


    }

    // Step 2. Receive INVITE Request

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ExpectedCalleeUri, v_MessageHeader_Invite, p_WaitDuration, -, p_GeolocationInfoAvailable, -, -, p_OIR);

    if (isvalue(v_IMS_DATA_REQ)) {

      v_IMS_DATA_REQ_Value := valueof(v_IMS_DATA_REQ);

      v_InviteRequest := v_IMS_DATA_REQ_Value.Request.Invite;

      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_MessageTemplate, p_GeolocationInfoAvailable);

      // check media description:

      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], p_AudioCodecList, v_CheckForbiddenCodecParameters);  /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_EVS_16000_1, f_IMS_SDP_CheckMediaDescr_AMR_16000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@

                                                                                                                                 @sic R5-172931: f_IMS_SDP_CheckMediaDescr_AudioCodecs instead of f_IMS_SDP_CheckMediaDescr_AudioCodec sic@

                                                                                                                                 @sic R5-180231: v_CheckForbiddenCodecParameters sic@ */

      v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Value, v_SDP_Message);

    }

    return v_InviteRequestWithSdp;

  }


Change 5 – f_IMS_MOCallSetup_Video_Step2
	  function f_IMS_MOCallSetup_Video_Step2(VideoCallTypeMO_Type p_VideoCallType,

                                         template SipUrl p_OtherCalleeUri := omit) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var template (present) SDP_Message v_SDP_Message_Step2;

    var INVITE_A_2_1_Context_Type v_Context;

    var IMS_DATA_REQ v_IMS_INVITE_REQ;

    var SDP_Message v_SDP_Message;

    var boolean v_A11 := true;                              /* @sic R5-176371: INVITE for creating a video call sic@ */


    var boolean v_PreconditionSupported := true;

    var OptionTag_List v_AdditionalOptionTagsForSupported := f_IMS_AdditionalOptionTagsForSupported(v_PreconditionSupported);   /* @sic R5-170781 sic@ */

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;




      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;


    select (p_VideoCallType) {

      case (SetupVideoCall) {

        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAudioVideo_RX(v_AudioCodecList);

        v_Context := A_2_1_A4;

      }

      case (AddVideo) {

        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAddVideo_RX();

        v_Context := A_2_1_A5;

      }

    }

    // Step 2. Receive INVITE Request

    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(v_Context, v_AdditionalOptionTagsForSupported, p_OtherCalleeUri, -, -, v_A11);  /* @sic R5s150180: p_A10 removed sic@ @sic R5-176209: v_A12 removed sic@ */

    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_Message_Step2);

    // check media description:

    if (p_VideoCallType == SetupVideoCall) {                                                 /* @sic R5-172931: no check of voice codecs for 17.1 sic@ */

      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], v_AudioCodecList);  /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_AMR_8000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@

                                                                                                @sic R5-168032, R5-168141: AMR_16000 sic@

                                                                                                @sic R5-172931 sic@ */

    }

    f_IMS_SDP_CheckMediaDescr_H264_90000(v_SDP_Message.media_list[1]);

    f_IMS_SDP_CheckMediaDescr_TcapPcfg(v_SDP_Message.media_list[1]);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);

  }


Change 6 – f_IMS_BuildSDP_InitialOfferAudio_RX
	  function f_IMS_BuildSDP_InitialOfferAudio_RX(boolean p_PreconditionsRequired := true,

                                               IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC return template (present) SDP_Message

  {

    var template (present) SDP_attribute_list v_SDP_MediaAttributes := {};

    var template (present) integer v_BandwidthRS := ?;               /* @sic R5-160906 sic@ */

    var template (present) integer v_BandwidthRR := (1..infinity);   /* @sic R5-160906 sic@ */

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes := {};


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();  /* @sic R5-171476: C.21a sic@ */

    var boolean v_IsCatM1 := f_IMS_TestcaseIsCatM1();   /* @sic R5-174483: C.21d sic@ */

    v_SDP_MediaAttributes := f_SDP_MediaAttributesAudioCodecListRX(p_AudioCodecList);   /* @sic R5-168031, R5-168053, R5-168054, R5-172931 sic@ */

    /* @sic R5-150709: cr_SDP_Attribute_inactive removed sic@ */

    if (not v_IsWLAN and not v_IsCatM1) {                                                       /* Attributes for media security mechanism may be required for C.21 only */

      v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_AddMediaSecurityAttributesRX(v_SDP_MediaAttributes);    /* @sic R5-160293,  R5-160915 new TC H.12.3 sic@ */

    }

    if (p_PreconditionsRequired) { /* @sic R5s150704, R5s150751 - Additional changes sic@ */

      if (v_IsWLAN) { /* @sic R5-171476 - WLAN Annex C.21a sic@ */

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_sendrecv, c_none, c_mandatory, c_optional);

      } else {

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_optional);

      }

    }

    return f_IMS_BuildSDP_RX(cr_SDP_Time_Any, cr_SDP_Media_Audio, v_SDP_MediaAttributes, v_SDP_PrecondionAttributes, -, v_BandwidthRS, v_BandwidthRR);

  }


Change 7 – f_IMS_BuildSDP_InitialAnswerAudio_TX
	  function f_IMS_BuildSDP_InitialAnswerAudio_TX(SDP_Message p_SDP_MessageInvite,

                                                boolean p_PreconditionsRequired := true,

                                                IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                                integer p_BandwidthAS := 37)

    runs on IMS_PTC

    return template (value) SDP_Message

  {

    var template (omit) SDP_attribute v_Attribute;

    var SDP_media_desc v_SDP_Media_Invite := p_SDP_MessageInvite.media_list[0];

    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParameters := f_SDP_FmtpDefaultParametersAudioCodecRX(p_AudioCodec);   /* @sic R5s160379 change 2 sic@ */

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(v_SDP_Media_Invite, p_AudioCodec, v_FmtpParameters);

    var integer v_BandwidthRS := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RS");

    var integer v_BandwidthRR := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RR");

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Media := cs_SDP_Bandwidth_List_Media(p_BandwidthAS, v_BandwidthRS, v_BandwidthRR); // @sic R5-123505, R5-135020; R5-144700: AS=p_BandwidthAS sic@

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes;

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes := omit;

    var template (omit) SemicolonParam_List v_FmtpParameterList_Answer := omit;


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();  /* @sic R5-171477 sic@ */

    var boolean v_IsCatM1 := f_IMS_TestcaseIsCatM1();   /* @sic R5-174483: C.21d sic@ */

    var SemicolonParam_List v_FmtpParameterList_Offer;

    var CharStringList_Type v_FmtpParametersToBoCopied;

    var charstring v_BwVal;

    select (p_AudioCodec) {                     // @sic R5-151968, R5-153999 sic@

      case (EVS_16000){

        if (f_IMS_PTC_RanTypeIsNR5GC()) {       // @sic R5-206382: implementation of TC_7_4 sic@

          v_FmtpParameterList_Offer := f_SDP_MediaDescr_GetFirstEVSCodecParameters(v_SDP_Media_Invite);

          if (match(v_FmtpParameterList_Offer, cr_EVS_Fmtp_AudioParamsBrBw("13.2", "swb"))) {

            v_BwVal := "13.2";                  // 34.229-5 A.4.1/2 183 Session Progress, NOTE 8

          } else {

            v_BwVal := "5.9-13.2";              // 34.229-5 A.4.1/2 183 Session Progress, NOTE 9

          }

          v_FmtpParameterList_Answer := { cs_EVS_Br(v_BwVal), cs_EVS_Bw("swb"), cs_Fmtp_ModesetParam("0,1,2"), cs_AMR_MaxRed };

        } 

        else {                                  // C.44 @sic R5-153999 sic@

          v_Attribute := f_SDP_AttributeList_GetAttribute(v_SDP_Media_Invite.attributes, cr_SDP_Attribute_fmtp(v_FmtAudio));

          v_BwVal := f_SIP_SemicolonParamList_GetParamValue(v_Attribute.fmtp.params.paramList, "bw");   /* @sic R5s170336/R5-172069 sic@ */

          v_FmtpParameterList_Answer := {cs_EVS_Bw(v_BwVal), cs_EVS_BwSend(v_BwVal), cs_EVS_BwRecv(v_BwVal), cs_AMR_MaxRed};   /* @sic R5s160379: Additional Change by MCC160 sic@

                                                                                                                                  @sic R5s170336/R5-172069: v_BwVal sic@ */

          v_FmtpParametersToBoCopied := { "br", "br-send", "br-recv" };

          v_FmtpParameterList_Offer := f_SDP_MediaDescr_GetCodecParameters(v_SDP_Media_Invite.attributes, v_FmtAudio);

          v_FmtpParameterList_Answer := f_SIP_SemicolonParamList_CopyParams(v_FmtpParameterList_Offer, v_FmtpParameterList_Answer, v_FmtpParametersToBoCopied);

        }

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_EVS_AudioDef(v_FmtAudio, valueof(v_FmtpParameterList_Answer));   // valueof to avoid compiler warning

      }

      case (AMR_16000){                         // @sic R5-153950 sic@

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AMR_WB_AudioDef(v_FmtAudio);  /* @sic R5s160379 change 2 sic@ */

      }

      case (AMR_8000){

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AudioDef(v_FmtAudio);   /* @sic R5-144693: new function f_SDP_MediaAttributes_AudioDef sic@ */

      }

    }

    if (not v_IsWLAN and not v_IsCatM1) {       // @sic R5-171476, R5-174483: not required for C.21a and C.21d sic@

      v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_CopyEcnCapabilityAttributesTX(v_SDP_Media_Invite, v_SDP_MediaAttributes);

    }

    if (not v_IsWLAN and not v_IsCatM1) {                                                     /* Attributes for media security mechanism may be required for C.21 only */

      v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_AddMediaSecurityAttributesTX(v_SDP_MediaAttributes);  /* @sic R5-160293,  R5-160915 new TC H.12.3 sic@ */

    }

    if (p_PreconditionsRequired) {  /* @sic R5s150704, R5s150751 - Additional changes sic@ */

      if (v_IsWLAN) {               /* @sic R5-171477 - WLAN Annex C.21a sic@ */

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);    // @sic R5s170904 change 3: sendrecv for remote precondition sic@

      } else {

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes := f_SDP_Attributes_Add_TX(v_SDP_PrecondionAttributes, cs_SDP_Attribute_conf_qos);   // @sic R5s120530 change 3.5 sic@

      }

    }

    return f_IMS_BuildSDP_TX(cs_SDP_Media_Audio(v_FmtAudio), v_Bandwidth_List_Media, v_SDP_MediaAttributes, v_SDP_PrecondionAttributes);

  }


Change 8 – f_IMS_BuildSDP_InitialOfferAudioVideo_RX
	  function f_IMS_BuildSDP_InitialOfferAudioVideo_RX(IMS_SDP_AudioCodecList_Type p_AudioCodecList) runs on IMS_PTC return template (present) SDP_Message

  {

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes;

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();  /* @sic R5-171478 sic@ */





      if (v_IsWLAN) {                      // C.25a

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_sendrecv, c_none, c_mandatory, c_optional);

      } else {                             // C.25

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_optional);

      }


    v_SDP_MediaAttributes_Audio := f_SDP_MediaAttributesAudioCodecListRX(p_AudioCodecList); // @sic R5-168032, R5-172931 sic@

    v_SDP_MediaAttributes_Video := {

      // The tcap/pcfg attributes need to be checked later when we know whether we have RTP/AVP or RTP/AVPF

      cr_SDP_Attribute_rtpmap(-, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(-, cr_Video_Fmtp_DefaultParametersMO)

    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                        cr_SDP_Media_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes,

                                        cr_SDP_Media_Video(-, (c_rtpAvp, c_rtpAvpf)),

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes);

  }


Change 9 – f_IMS_BuildSDP_InitialOfferAddVideo_RX
	  function f_IMS_BuildSDP_InitialOfferAddVideo_RX() runs on IMS_PTC return template (present) SDP_Message

  {

    var template (present) SDP_attribute_rtpmap_codec v_AudioCodec;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (present) charstring v_StrengthRemote := (c_optional, c_mandatory);  /* @sic R5-165081: (c_optional, c_mandatory) sic@ */

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video := {

      // The tcap/pcfg attributes need to be checked later when we know whether we have RTP/AVP or RTP/AVPF

      cr_SDP_Attribute_rtpmap(-, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(-, cr_Video_Fmtp_DefaultParametersMO)

    };

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();








      v_AudioCodec := cr_RTPMAP_AMR_16000;

      v_SDP_PrecondionAttributes_Audio := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, v_StrengthRemote);

      if (v_IsWLAN) {

        v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_sendrecv, c_none, c_mandatory, v_StrengthRemote);

      }

      else {

        v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, v_StrengthRemote);

      }


    v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(-, v_AudioCodec), /* @sic R5-168141 sic@ */

      cr_SDP_Attribute_fmtp                     /* => any parameters for audio */

    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                        cr_SDP_Media_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        cr_SDP_Media_Video(-, (c_rtpAvp, c_rtpAvpf)),

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }


Change 10 – f_IMS_BuildSDP_InitialAnswerAudioVideo_TX
	  function f_IMS_BuildSDP_InitialAnswerAudioVideo_TX(VideoCallTypeMO_Type p_VideoCallType,

                                                     SDP_Message p_SDP_Message_Step2)

    runs on IMS_PTC return template (value) SDP_Message

  {

    var SDP_media_desc v_SDP_Media_AudioRx := p_SDP_Message_Step2.media_list[0];

    var SDP_media_desc v_SDP_Media_VideoRx := p_SDP_Message_Step2.media_list[1];

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Audio := v_SDP_Media_AudioRx.bandwidth;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Video := v_SDP_Media_VideoRx.bandwidth;

    var template (omit) SDP_attribute v_FmtpAttributeVideo;

    var template (value) SemicolonParam_List v_FmtpParamListVideo := {};

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes_Audio := {};

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes_Video := {};

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes_Audio := omit;

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes_Video := omit;

    var charstring v_FmtAudio;

    var charstring v_FmtVideo;

    var boolean v_TcapPcfg;

    var integer v_SessionBandwidthAS;


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    /***** media attributes for audio *****/






      v_FmtAudio := f_SDP_MediaDescr_GetAMR_WB_16000_1(v_SDP_Media_AudioRx); // @sic R5-168032, R5-168119, R5-168141 sic@

      v_SDP_MediaAttributes_Audio := f_SDP_MediaAttributes_AMR_WB_AudioDef(v_FmtAudio);   /* @sic R5-144693: new function f_SDP_MediaAttributes_AudioDef sic@

                                                                                             @sic R5-168032, R5-168119, R5-168141 sic@

                                                                                             @sic R5s170427: f_SDP_MediaAttributes_AMR_WB_AudioDef instead of f_SDP_MediaAttributes_AMR_WB_AudioCommon sic@ */


    if ((p_VideoCallType == SetupVideoCall) and not v_IsWLAN) {  /* NOTE: there is nothing specified in 17.1 or C.25a regarding "a=inactive" */

      if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_AudioRx.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-151796/R5-151797 sic@ */

        v_SDP_MediaAttributes_Audio := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Audio, cs_SDP_Attribute_inactive);  /* in case of C.25b this does not make sense, nevertheless it does not harm

                                                                                                                            as long as the UE does not provide a=inactive at step 2 */

      }

    }

    /***** media attributes for video *****/

    v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(v_SDP_Media_VideoRx);

    v_TcapPcfg := fl_IMS_Video_TcapPcfgRequired(v_SDP_Media_VideoRx);

    v_FmtpAttributeVideo := f_SDP_AttributeList_GetAttribute(v_SDP_Media_VideoRx.attributes, cr_SDP_Attribute_fmtp(v_FmtVideo));   // @sic R5s141336 change 2.2 - MCC160 implementation sic@ */

    if (isvalue(v_FmtpAttributeVideo)) {

      v_FmtpParamListVideo :=  valueof(v_FmtpAttributeVideo.fmtp.params.paramList);      /* @sic R5-150734, R5-150736 sic@ */

    }

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtpmap(v_FmtVideo, cs_RTPMAP_H264_90000));

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_fmtp(v_FmtVideo, cs_Fmtp_ParamList(v_FmtpParamListVideo)));

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_acfg(1, "t=1"), v_TcapPcfg);

    if ((p_VideoCallType == SetupVideoCall) and not v_IsWLAN) {  /* NOTE: there is nothing specified in 17.1 or C.25a or C.25b regarding "a=inactive" */

      if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_VideoRx.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-150694 sic@ */

        v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_inactive);

      }

    }

    /***** precondition attributes *****/


      if (v_IsWLAN) {                // C.25a: all resource are available

        v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes_Video := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);

      } else {                       // C.25 or 17.1

        v_SDP_PrecondionAttributes_Video := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_PrecondionAttributes_Video, cs_SDP_Attribute_conf_qos);

        select (p_VideoCallType) {

          case (SetupVideoCall) {    // C.25

            v_SDP_PrecondionAttributes_Audio := v_SDP_PrecondionAttributes_Video;   // same preconditions for audio and video

          }

          case (AddVideo) {          // 17.1

            v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);

          }

        }

      }


    if (v_IsWLAN and (p_VideoCallType == SetupVideoCall)) {

      v_SessionBandwidthAS := f_SDP_BandwidthList_GetBandwidth(p_SDP_Message_Step2.bandwidth, "AS");    /* C.25a: "b=AS: (bandwidth-value) [Note 1]" */

    } else {    /* @sic R5-174556: b=AS:30 for C.25b (same as for C.25) sic@ */

      v_SessionBandwidthAS := 30;

    }

    return f_IMS_BuildSDP_AudioVideo_TX(v_SessionBandwidthAS,

                                        cs_SDP_Media_Audio(v_FmtAudio),

                                        v_Bandwidth_List_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        cs_SDP_Media_Video(v_FmtVideo),

                                        v_Bandwidth_List_Video,

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }


Change 11 – f_IMS_BuildSDP_MTCallAudioVideo_TX
	  function f_IMS_BuildSDP_MTCallAudioVideo_TX(VideoCallTypeMT_Type p_VideoCallType,

                                              VideoCallMT_MessageTX_Type p_MessageType,

                                              charstring p_DirectionRemote := c_none)

    runs on IMS_PTC

    return template (value) SDP_Message

  {

    var IMS_SDP_AudioCodecList_Type v_AudioCodecs;

    var template (value) IMS_SDP_MediaDescriptionRtpList_Type v_MediaDescriptionRtpList;

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (value) SDP_fmt_list v_FmtListAudio;

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Audio := cs_SDP_Bandwidth_List_Media(37, 0, 2000);   // @sic R5-151956, R5s170056/R5-171489: 37 instead of 30 sic@

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Video := cs_SDP_Bandwidth_List_Media(315, 0, 2500);

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes_Audio := omit;

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes_Video := omit;

    var integer v_SessionBandwidthAS := 352;            /* @sic R5s170056/R5-171489/R5-174557 sic@ */

    var charstring v_FmtVideo := tsc_SDP_FmtVideo;      /* @sic R5-197050: tsc_SDP_FmtVideo instead of fl_IMS_BuildSDP_FmtVideo sic@ */

    var template (value) SDP_media_field v_MediaVideo := cs_SDP_Media_Video(v_FmtVideo);


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    var boolean v_IsVideoCallSetup := (p_VideoCallType == SetupVideoCall);

    var boolean v_IsUpdate := (p_MessageType == Update);

    var boolean v_IsNotRemoveVideo := (p_VideoCallType != RemoveVideo);

    if (v_IsVideoCallSetup) {           /* C.26/C.26a/C.26b */





        if (not v_IsUpdate) {           /* C.21/C.21a INVITE (Step 1) */

          v_AudioCodecs := { AMR_16000, DTMF_16000, AMR_8000, DTMF_8000 };

          if (v_IsWLAN) {               /* C.21a INVITE (Step 1) */

            v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_none, c_mandatory, c_optional);

          } else {                      /* C.21 INVITE (Step 1) */

            v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_optional);

          }

        } else {                        /* C.26 UPDATE (Step 7) */

          v_AudioCodecs := { AMR_16000 };

          v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, p_DirectionRemote, c_mandatory, c_mandatory);

        }

        v_SDP_PrecondionAttributes_Video := v_SDP_PrecondionAttributes_Audio;   // same precondions for video and audio


    }

    else {                              /* Add/remove video to/from call */





        v_AudioCodecs := { AMR_16000 };


      if (v_IsNotRemoveVideo) {


          if (not v_IsUpdate) {         /* 17.2 INVITE (Step 1) */

            v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_optional);

            v_SDP_PrecondionAttributes_Video := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_optional);

          } else {                      /* 17.2 UPDATE (Step 6) */

            v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);

            v_SDP_PrecondionAttributes_Video := cs_SDP_PrecondionAttributes(c_sendrecv, p_DirectionRemote, c_mandatory, c_mandatory);

          }


      } else {                          /* Remove video: 17.2 step 10, H.17.2 step 9 */

        v_SessionBandwidthAS := 37;     /* audio only @sic R5-171489 sic@ */

        v_MediaVideo.ports.port_number := 0;


          v_SDP_PrecondionAttributes_Audio := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_optional);

          v_SDP_PrecondionAttributes_Video := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_optional);


      }

    }

    // ***** Media description audio *****

    v_MediaDescriptionRtpList := f_SDP_MediaDescriptionAudioCodecListTX(v_AudioCodecs); /* @sic R5-172930 sic@ */

    v_FmtListAudio := v_MediaDescriptionRtpList.FmtList;

    v_SDP_MediaAttributes_Audio := v_MediaDescriptionRtpList.AttributeList;

    if (not (v_IsUpdate and (p_VideoCallType == AddVideo))) { // @sic R5-151956: there are no ptime, maxptime for UPDATE of 17.2 sic@

      // additional attributes for audio: ptime, maxptime

      v_SDP_MediaAttributes_Audio := f_SDP_Attributes_Concat_TX(v_SDP_MediaAttributes_Audio, cs_SDP_MediaAttributes_ptime_maxptime);

    }

    // ***** Media description video *****

    v_SDP_MediaAttributes_Video := {

      cs_SDP_Attribute_rtpmap(v_FmtVideo, cs_RTPMAP_H264_90000),

      cs_SDP_Attribute_fmtp(v_FmtVideo, cs_Fmtp_ParamList(cs_Fmtp_VideoParamsDef))

    };

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtcp_fb("*", "trr-int 5000"), v_IsNotRemoveVideo);  /* @sic R5s141329 change 2.2: attribute is for video sic@ */

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtcp_fb("*", "nack"),         v_IsNotRemoveVideo);  /* @sic R5s141329 change 2.2: attribute is for video sic@ */

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtcp_fb("*", "nack pli"),     v_IsNotRemoveVideo);  /* @sic R5s141329 change 2.2: attribute is for video sic@ */

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtcp_fb("*", "ccm fir"),      v_IsNotRemoveVideo);  /* @sic R5s141329 change 2.2: attribute is for video sic@ */

    v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes_Video, cs_SDP_Attribute_rtcp_fb("*", "ccm tmmbr"),    v_IsNotRemoveVideo);  /* @sic R5s141329 change 2.2: attribute is for video sic@ */

    return f_IMS_BuildSDP_AudioVideo_TX(v_SessionBandwidthAS,

                                        cs_SDP_Media_MultiAudio(v_FmtListAudio),

                                        v_Bandwidth_List_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        v_MediaVideo,

                                        v_Bandwidth_List_Video,

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }


Change 12 – f_IMS_BuildSDP_MTCallAudioVideo_RX
	  function f_IMS_BuildSDP_MTCallAudioVideo_RX(VideoCallTypeMT_Type p_VideoCallType,

                                              template (omit) VideoCallMT_MessageRX_Type p_MessageType := omit)

    runs on IMS_PTC

    return template (present) SDP_Message

  {

    var template (present) SDP_attribute_rtpmap_codec v_AudioCodec;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio := {}; // @sic R5-170782 sic@

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video := {}; // @sic R5-170782 sic@

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Final := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);  /* preconditions when resources are fully available */

    var charstring v_FmtAudio;

    var charstring v_FmtVideo := tsc_SDP_FmtVideo;               /* @sic R5-197050: tsc_SDP_FmtVideo instead of fl_IMS_BuildSDP_FmtVideo sic@ */


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();






      v_AudioCodec := cr_RTPMAP_AMR_16000;                       // @sic R5-168058; R5-168142 sic@

      v_FmtAudio := f_SDP_AudioCodec_GetFmt(AMR_16000);          /* @sic R5-197050: fix mapping of fmt values sic@ */

      if (not v_IsWLAN and match(Response183, p_MessageType)) {  // i.e. 183 Session Progress (C.26 step 4 or 17.2 step 3)

        v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes((c_none, c_sendrecv), c_none, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes_Video := f_SDP_Attributes_Add_RX(v_SDP_PrecondionAttributes_Video, cr_SDP_Attribute_conf_qos);

        if (p_VideoCallType == SetupVideoCall) {                 // C.26 step 4: same preconditions for video and audio

          v_SDP_PrecondionAttributes_Audio := v_SDP_PrecondionAttributes_Video;

        } else {                                                 // 17.2 step 3: audio is there already

          v_SDP_PrecondionAttributes_Audio := v_SDP_PrecondionAttributes_Final;

        }

      } else {  // WLAN or 200 OK

        v_SDP_PrecondionAttributes_Audio := v_SDP_PrecondionAttributes_Final;

        v_SDP_PrecondionAttributes_Video := v_SDP_PrecondionAttributes_Final;

      }


    v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_AudioCodec),

      cr_SDP_Attribute_fmtp(v_FmtAudio)

    };

    v_SDP_MediaAttributes_Video := {

      cr_SDP_Attribute_rtpmap(v_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(v_FmtVideo, cr_Video_Fmtp_DefaultParametersMT)  /* @sic R5s141329 - MCC160 implementation sic@

                                                                               @sic R5-150703: same parameters for C.26 and 17.2 sic@ */

    };

    return f_IMS_BuildSDP_AudioVideo_RX(-,

                                        cr_SDP_Media_Audio(v_FmtAudio),

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        cr_SDP_Media_Video(v_FmtVideo),

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }


Change 13 – f_IMS_CC_MO_SpeechCall
	  function f_IMS_CC_MO_SpeechCall(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,

                                  template (omit) IMS_SDP_AudioCodec_Type p_AudioCodecToBeSelected := omit,

                                  boolean p_PreconditionsRequired := true) runs on IMS_PTC return INVITE_Request

  {

    var IMS_CC_MO_SpeechCallInfo_Type v_MO_SpeechCallInfo;      /* @sic R5-197039, R5-197040: IMS_CC_MO_SpeechCallInfo_Type instead of IMS_InviteRequestWithSdp_Type sic@ */

    var INVITE_Request v_InviteRequest;

    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected;

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();


    if (v_IsWLAN) {

      v_InviteRequest := f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b(IPCAN_MO_SpeechCall, p_AudioCodecList, p_PreconditionsRequired);

    }

    else {

      if (ispresent(p_AudioCodecToBeSelected)) {

        v_AudioCodecToBeSelected := valueof(p_AudioCodecToBeSelected);

      } else {

        v_AudioCodecToBeSelected := p_AudioCodecList[0];

      }

      v_MO_SpeechCallInfo := f_IMS_MOCallSetup_AnnexC21C21b(p_AudioCodecList, v_AudioCodecToBeSelected);

      v_InviteRequest := v_MO_SpeechCallInfo.InviteRequestWithSdp.Invite;

    }

    return v_InviteRequest;

  }


Change 14 – f_IMS_MTCallwithTIR
	  function f_IMS_MTCallwithTIR(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_Response_180_183_WithSdp_Type v_Response_180_183_WithSdp;  // @sic R5s180436: IMS_Response_180_183_WithSdp_Type instead of v_OptionalResponseWithSdp sic@

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionProgress183;

    var template (present) SDP_Message v_SDP_Message;

    var MessageHeader v_MessageHeader_SessionProgress;

    var template (omit) MessageHeader v_MessageHeader_Ringing := omit;

    var MessageHeader v_MessageHeader_200OK;

    var SipUrl v_ContactUrl;


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    var boolean v_PreconditionsRequired := true;


    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected := p_AudioCodecList[0];

    timer t_Dummy;    // according to the test case prose of G.15.4 and H.15.4 the 180 Ringing seems to be optional

    f_IMS_TestBody_Set(true);

    f_UT_MMI_ActivateIMS_TIR(MMI);

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_BuildInviteRequestWithSdp(p_AudioCodecList, v_PreconditionsRequired);

    if (v_IsWLAN) {
      // @siclog "Step 1" siclog@

      IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequestWithSdp.Invite));

      // @siclog "Step 2 - 5" siclog@

      v_SDP_Message := f_IMS_BuildSDP_AnswerAudio_RX(FinalAnswer, v_AudioCodecToBeSelected, v_PreconditionsRequired);

      v_Response_180_183_WithSdp := f_IMS_MTCallSetup_AnnexC11aC11cC26aC26b_Steps2_5(t_Dummy, v_InviteRequestWithSdp, v_SDP_Message);

      f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  /* trigger IPCAN to do the routing for RTP/RTCP loopback */

      if (ispresent(v_Response_180_183_WithSdp.Response180)) {

        v_MessageHeader_Ringing := v_Response_180_183_WithSdp.Response180.msgHeader;

      }

      if(ispresent (v_Response_180_183_WithSdp.Response183)){ // @sic R5-184056 sic@

        v_MessageHeader_SessionProgress :=  v_Response_180_183_WithSdp.Response183.msgHeader;

        fl_MessageHeader_CheckPrivacy(v_MessageHeader_SessionProgress);

      }

    } else {

      // @siclog "Step 1 - 4" siclog@

      v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp, v_AudioCodecToBeSelected, v_PreconditionsRequired);

      v_SessionProgress183 := v_ResponseWithSdp.Response;

      v_MessageHeader_SessionProgress := v_SessionProgress183.msgHeader;

      fl_MessageHeader_CheckPrivacy(v_MessageHeader_SessionProgress);

      // Step 5 - 6. Send PRACK, Receive 200 OK

      f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequestWithSdp.Invite, v_SessionProgress183.msgHeader);     /* @sic R5-181141: steps 5-6 before sending the trigger sic@ */

      // Establish Dedicated EPS bearer

      // See TS. 36.508 EUTRA/EPS signalling for IMS MT speech call Table 4.5A.7.3-1 Step 9-12

      f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);

      // Step 7 - 11. Send UPDATE, Receive 200 OK, Receive optional 180 Ringing and send PRACK if requested

      v_MessageHeader_Ringing := f_IMS_MTCallSetup_AnnexC11_Steps7_11(v_InviteRequestWithSdp, v_ResponseWithSdp, v_AudioCodecToBeSelected);

    }

    if (ispresent(v_MessageHeader_Ringing)) {

      fl_MessageHeader_CheckPrivacy(valueof(v_MessageHeader_Ringing));

    }

    // Step 11A - Make UE accept the speech AMR offer

    f_UT_AnswerCall(MMI);

    // Step 12. Receive 200 OK, Send ACK

    v_ResponseWithSdp := f_IMS_MTCallSetup_Receive200OK_SendACK(v_InviteRequestWithSdp);

    v_MessageHeader_200OK := v_ResponseWithSdp.Response.msgHeader;

    fl_MessageHeader_CheckPrivacy(v_MessageHeader_200OK);

    // MT Call is now established

    f_Delay(3.0);    /* @sic R5s130510 change 1; draft prose CR sic@ */

    // Step 14-15. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    // @siclog "Step 14 -15" siclog@

    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog); //@sic R5s180433 sic@

    f_IMS_CallReleaseMT(v_ContactUrl);

    f_UT_MMI_DeactivateIMS_TIR(MMI);

    f_IMS_TestBody_Set(false);

  }


Change 15 – fl_IMS_BuildSDP_MOCallAudioVideo_ReleaseVideo
	  function fl_IMS_BuildSDP_MOCallAudioVideo_ReleaseVideo(SDP_Message p_OriginalSdpOffer) runs on IMS_PTC return template (present) SDP_Message

  {

    var charstring v_FmtAudio;

    var charstring v_FmtVideo;

    var template (present) SDP_attribute_rtpmap_codec v_RtpmapAttrAudio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes;








      v_FmtAudio := f_SDP_MediaDescr_GetAMR_WB_16000_1(p_OriginalSdpOffer.media_list[0]); /* @sic R5s170012/R5-171474: f_SDP_MediaDescr_GetAMR_WB_16000_1 sic@ */

      v_RtpmapAttrAudio := cr_RTPMAP_AMR_16000;   /* @sic R5s170012/R5-171474: cr_RTPMAP_AMR_16000 sic@ */

      v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory)); /* @sic R5s160604 change 3 sic@ */


    v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(p_OriginalSdpOffer.media_list[1]);

    v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_RtpmapAttrAudio),

      cr_SDP_Attribute_fmtp(v_FmtAudio)

    };

    v_SDP_MediaAttributes_Video := {

      cr_SDP_Attribute_rtpmap(v_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(v_FmtVideo)

    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                        cr_SDP_Media_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes,

                                        cr_SDP_Media(c_video, 0, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(v_FmtVideo)),  /* port number == 0 => release video stream

                                                                                                                        @sic R5s160604: RTP/AVPF sic@ */

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes);

  }


Change 16 – fl_IMS_MOCall_ReleaseVideo_Invite
	  function fl_IMS_MOCall_ReleaseVideo_Invite(SDP_Message p_OriginalSdpOffer) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var IMS_DATA_REQ v_IMS_INVITE_REQ;

    var SDP_Message v_SDP_Message;

    var boolean v_A11 := false;                        /* @sic R5-176371: A11 - INVITE for creating a video call => false in case of call release sic@ */


    var boolean v_PreconditionSupported := true;

    var OptionTag_List v_AdditionalOptionTagsForSupported := f_IMS_AdditionalOptionTagsForSupported(v_PreconditionSupported);

    var template (present) SDP_Message v_SDP_MessageTemplate := fl_IMS_BuildSDP_MOCallAudioVideo_ReleaseVideo(p_OriginalSdpOffer);

    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A5, v_AdditionalOptionTagsForSupported, -, -, -, v_A11);  /* @sic R5-176209: v_A12 removed sic@ */

    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_MessageTemplate);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);

  }


Change 17 – f_IMS_MOSpeechAddVideoRemoveVideo
	  function f_IMS_MOSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {

    var template (present) SipUrl v_CalleeContactUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri);

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var INVITE_Request v_InviteRequest;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (value) SDP_Message v_SdpAnswer;

    var template (value) MessageBody v_MessageBody;

    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;

    var charstring v_CallIdentification;

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;

    var IMS_PreconditionIndication_Type v_PreconditionIndication;


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();






      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

      v_PreconditionIndication := PreconditionRequired;

      v_PreconditionsRequired := true;


    // Initial conditions - Setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, -, normalCall_ReturnId);        // @sic R5s160855 sic@

    f_IMS_CC_MO_SpeechCall(v_AudioCodecList, -, v_PreconditionsRequired);   // MO voice call according

    v_CallIdentification := f_UT_ReceivedIMSCallId(MMI);

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_UT_AddVideoToIMSCall(MMI, v_CallIdentification);               // @sic R5s160855 sic@

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(AddVideo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsWLAN) {    // G.17.1, H.17.1

      // @siclog "Step 3" siclog@

      f_IMS_MOCallSetup_Send100Trying(v_InviteRequestWithSdp);        // Send 100 Trying

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MO_AddVideo);

      // @siclog "Step 4 - 6 (H.17.1): Void" siclog@

      // @siclog "Step 4 (G.17.1)" siclog@

      // @siclog "Step 7 (H.17.1)" siclog@

      v_SdpAnswer := f_IMS_BuildSDP_InitialAnswerAudioVideo_TX(AddVideo, v_InviteRequestWithSdp.SdpOffer);    // same as for C.25a/C.25b

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, -, -, v_PreconditionIndication);

      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response, cs_MessageBody_SDP(v_SdpAnswer))));

      // @siclog "Step 5 (G.17.1)" siclog@

      // @siclog "Step 8 (H.17.1)" siclog@

      v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

      f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactUri, v_MessageHeader_Request)));

    }

    else {                        // 17.1

      // @siclog "Step 3 - 4" siclog@

      v_MessageHeader_Response := f_IMS_MOCallSetup_Video_Step3_4(AddVideo, v_InviteRequestWithSdp);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MO_AddVideo);

      // @siclog "Step 5 - 8" siclog@

      f_IMS_MOCallSetup_Video_Steps5_8(AddVideo, v_InviteRequestWithSdp, v_MessageHeader_Response);

      // @siclog "Step 9 - 10" siclog@

      f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);

    }

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have really been established

    // @siclog "Step 10a (17.1)" siclog@

    // @siclog "Step 6 (G.17.1)" siclog@

    // @siclog "Step 9 (H.17.1)" siclog@

    f_UT_IMS_RemoveVideoMedia(MMI, v_CallIdentification);

    // @siclog "Step 11 (17.1)" siclog@

    // @siclog "Step 7 (G.17.1)" siclog@

    // @siclog "Step 10 (H.17.1)" siclog@

    v_InviteRequestWithSdp := fl_IMS_MOCall_ReleaseVideo_Invite(v_InviteRequestWithSdp.SdpOffer);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    // @siclog "Step 12 (17.1)" siclog@

    // @siclog "Step 8 (G.17.1)" siclog@

    // @siclog "Step 11 (H.17.1)" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // @siclog "Step 12a (17.1)" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 13 (17.1)" siclog@

    // @siclog "Step 9 (G.17.1)" siclog@

    // @siclog "Step 12 (H.17.1)" siclog@

    v_MessageBody := cs_MessageBody_SDP(f_IMS_BuildSDP_CommonResponseToOfferTX(v_InviteRequestWithSdp.SdpOffer));

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, -, -, v_PreconditionIndication);  // @sic R5s170344: cs_ContentTypeSDP, PreconditionRequired sic@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response, v_MessageBody)));                         // @sic R5s170344: v_MessageBody sic@

    // @siclog "Step 14 (17.1)" siclog@

    // @siclog "Step 10 (G.17.1)" siclog@

    // @siclog "Step 13 (H.17.1)" siclog@

    v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactUri, v_MessageHeader_Request)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // @siclog "Step 15 - 16 (17.1)" siclog@

    // @siclog "Step 11 - 12 (G.17.1)" siclog@

    // @siclog "Step 14 - 15 (H.17.1)" siclog@

    f_IMS_CallReleaseMO(v_InviteRequest);

    f_IMS_TestBody_Set(false);

  }


Change 18 – f_IMS_MTSpeechAddVideoRemoveVideo
	  function f_IMS_MTSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {

    var IMS_CC_MT_SpeechCallInfo_Type v_SpeechCallInfo;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionProgress183;

    var template (present) IMS_Response_Type v_ExpectedSessionProgress183;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value)   SDP_Message v_SdpOffer;

    var template (present) SDP_Message v_SdpAnswer;

    var template SDP_Message v_SdpAnswer200OK;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (omit) SDP_attribute v_SDP_Attribute;

    var template (omit) IMS_DATA_RSP v_IMS_DATA_RSP_Optional;

    var SipUrl v_ContactUrl;

    var InternetProtocol_Type v_Protocol;

    var SDP_media_desc v_MediaDesc;

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;


    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    timer t_Timer := 5.0;

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);    // @sic R5s190003, R5s190006 sic@

    if (v_IsWLAN) {





        v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

        v_PreconditionsRequired := true;


      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11aC11c(v_AudioCodecList, v_PreconditionsRequired);

    }

    else {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11C11b(v_AudioCodecList);

    }

    v_ContactUrl := v_SpeechCallInfo.ContactUrl;

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // Step 1 SS sends re-INVITE with second SDP offer to add video.

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsWLAN) {

      // @siclog "Step 2" siclog@

      v_Protocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

      v_SdpAnswer200OK := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo);

      //@siclog "Step 2A" siclog@


        v_ExpectedSessionProgress183 := cr_Response(c_statusLine183,

                                                    f_IMS_InviteResponse_183_MessageHeaderRX(v_InviteRequest),

                                                    cr_MessageBody_SDP);

        //Start timer

        t_Timer.start; //not in Prose!!

        v_IMS_DATA_RSP_Optional := f_IMS_ReceiveResponse_OptionalTrying_Timeout(t_Timer, v_InviteRequest, v_Protocol, v_ExpectedSessionProgress183);

        t_Timer.stop;

        //@siclog "Step 2B-2C" siclog@

        if (isvalue(v_IMS_DATA_RSP_Optional)) { // @sic R5s190003 sic@

          v_SessionProgress183 := valueof(v_IMS_DATA_RSP_Optional.Response);

          f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_SessionProgress183.messageBody, v_SdpAnswer200OK);

          f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionProgress183.msgHeader);

          v_SdpAnswer200OK := omit;

        }




      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      //@siclog "Step 3 - 5: Void (H.17.2)" siclog@

    }

    else {

      // @siclog "Step 2: Void" siclog@

      // @siclog "Step 3" siclog@

      v_ResponseWithSdp := f_IMS_MTCallSetup_Video_Receive183(AddVideo, v_InviteRequestWithSdp);    /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_SessionProgress183 := v_ResponseWithSdp.Response;

      // @siclog "Step 4 - 5" siclog@

      f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionProgress183.msgHeader);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);            /* @sic R5-181141: trigger moved after step 4/5 sic@ */

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      // @siclog "Step 6" siclog@

      v_MediaDesc := valueof(f_IMS_SDP_MediaList_GetMediaStream(v_ResponseWithSdp.SdpMessage.media_list, c_video));         /* @sic R5s180304: f_IMS_SDP_MediaList_GetMediaStream sic@ */

      v_SDP_Attribute := f_SDP_AttributeList_GetAttribute(v_MediaDesc.attributes, cr_SDP_Attribute_curr_qos(c_local, ?));

      v_SdpOffer := f_IMS_BuildSDP_MTCallAudioVideo_TX(AddVideo, Update, valueof(v_SDP_Attribute.curr.direction)); /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(v_InviteRequest, PreconditionSupported);     /* @sic R5-160795: PreconditionSupported sic@ */

      v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();                                                    /* @sic R5s160540 change 1.5 sic@ */

      IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_SdpOffer))));

      // @siclog "Step 7" siclog@

      v_SdpAnswer := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, Response200);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cr_ContentTypeSDP);   /* @sic R5-160795, R5-162930: precondition required sic@ */

      v_IMS_DATA_RSP_Optional := f_IMS_ReceiveResponse(v_MessageHeader_Update, v_Protocol, cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));

      f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP_Optional.Response.messageBody, v_SdpAnswer);

      v_SdpAnswer200OK := omit;

    }

    // @siclog "Step 7A (17.2)" siclog@

    // @siclog "Step 2D (G.17.2)" siclog@

    // @siclog "Step 6 (17.2)" siclog@

    f_UT_AnswerCall(MMI);

    // @siclog "Step 8 - 9 (17.2)" siclog@

    // @siclog "Step 3 - 4 (G.17.2)" siclog@

    // @siclog "Step 7 - 8 (H.17.2)" siclog@

    // The UE responds to the re-INVITE with a 200 OK final response.

    // The SS acknowledges the receipt of 200 OK for the re-INVITE.

    f_IMS_MTCallSetup_Receive200OK_SendACK(v_InviteRequestWithSdp, v_SdpAnswer200OK);

    // @siclog "Step 10 (17.2)" siclog@

    // @siclog "Step 5 (G.17.2)" siclog@

    // @siclog "Step 9 (H.17.2)" siclog@

    // SS sends a re-INVITE with a SDP offer indicating that the video component is removed from the call

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    // @siclog "Step 10A - 11 (17.2)" siclog@

    // @siclog "Step 6 - 7 (G.17.2)" siclog@

    // @siclog "Step 10 - 11 (H.17.2)" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);    /* @sic R5-160795, R5-162930: cr_ContentTypeSDP, precondition required sic@ */

    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite,

                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,

                                         cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));  /* @sic R5s160842, R5s170013 sic@ */

    // @siclog "Step 11A (17.2)" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 12 (17.2)" siclog@

    // @siclog "Step 8  (G.17.2)" siclog@

    // @siclog "Step 12 (H.17.2)" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // @siclog "Step 13: Void (17.2)" siclog@

    // @siclog "Step 14 - 16 (17.2)" siclog@

    // @siclog "Step 9 - 10  (G.17.2)" siclog@

    // @siclog "Step 13 - 14 (H.17.2)" siclog@

    f_IMS_CallReleaseMT(v_ContactUrl);

    f_IMS_TestBody_Set(false);

  }


Change 19 – f_IMS_PTC_Security_GetHashAlgorithm
	  function f_IMS_PTC_Security_GetHashAlgorithm() runs on IMS_PTC return charstring

  {

    var charstring v_Algorithm;

    select (vc_IMS_Global.ImsInfo.SecurityScheme) {




      case (IMS_Security) {

        v_Algorithm := tsc_AKAv1_MD5;

      }

      case else {

        v_Algorithm := "";   // dummy

      }

    }

    return v_Algorithm;

  }


Change 20 – f_IMS_MOCallSetup_AnnexC21C25_Steps5_8
	  function f_IMS_MOCallSetup_AnnexC21C25_Steps5_8(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                  template (value) MessageHeader p_MessageHeader_Prev183Response,

                                                  boolean p_GeolocationInfoAvailable := false,

                                                  template SDP_Message p_SDP_Message_Step5or7 := omit,

                                                  template (omit) SDP_Message p_PrevSDP_Message := omit,

                                                  IMS_CallType_Type p_CallType := NORMAL_CALL,

                                                  boolean p_ResourcesEstablished := false,

                                                  IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                                  IMS_PreconditionIndication_Type p_PreconditionIndicationStep5and7 := PreconditionRequired) runs on IMS_PTC return SDP_Message

  {

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_Prev183Response);    /* @sic R5s130333 change 8.2 - acc. to A.2.4, A.2.5 same as in previous response sic@ */

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_Message := p_InviteRequestWithSdp.SdpOffer;  /* @sic R5w140112 sic@ */

    var template (present) SDP_Message v_SDP_Message_Step5or7;

    var boolean v_MediaConfigured := false;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var template (omit) MessageBody v_MessageBodyTx := omit;

    var template (omit) ContentType v_ContentType := omit;

    var PRACK_Request v_PrackRequest;

    var SDP_Message v_SDP_LatestOffer;


    if (isvalue(p_PrevSDP_Message)) {   /* e.g. in case of C.27 the INVITE's SDP message is not the latest SDP sent by the UE */

      v_SDP_Message := valueof(p_PrevSDP_Message);

    }

    if (not ispresent(p_SDP_Message_Step5or7)) {

      v_SDP_Message_Step5or7 := f_IMS_BuildSDP_UpdatedOfferAudio_RX(v_SDP_Message, p_ResourcesEstablished, p_AudioCodec);  /* => use speech per default */

    } else {

      v_SDP_Message_Step5or7 := p_SDP_Message_Step5or7;         /* In case of video the SDP message needs to be handed over

                                                                   NOTE: compiler warning cannot be avoided here */

    }

    // Step 5. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, p_MessageHeader_Prev183Response);

    v_IMS_DATA_REQ := f_IMS_ReceiveRequest(car_IMS_Prack_Request(cr_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack, cr_MessageBody_SDP ifpresent)));

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // fl_IMS_Params_CheckSigComp(v_PrackRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 4 (PRACK) "comp=sigcomp" is optional for Via and Route

    if (ispresent(v_PrackRequest.messageBody)) {

      f_IMS_MessageHeader_CheckPreconditionIndicationForOffer(v_PrackRequest.msgHeader, p_PreconditionIndicationStep5and7);   /* @sic R5-160794 sic@ */

      v_SDP_LatestOffer := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_PrackRequest.messageBody, v_SDP_Message_Step5or7);  // @sic R5w140112 sic@

      // check session version increased @sic R5-150702: done in f_IMS_SIP_DecodeMatchAndCheckSDP sic@ */

      v_ContentType := cs_ContentTypeSDP;

      v_MessageBodyTx := cs_MessageBody_SDP(f_IMS_BuildSDP_UpdatedAnswer_TX(v_SDP_LatestOffer));    /* @sic R5w140112 sic@ */

      v_MediaConfigured := true;

    }

    // Step 6. Send 200 OK

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL,

                                     cs_Response(c_statusLine200,

                                                 f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader, v_ContentType, -, -, PreconditionRequired, p_CallType),

                                                 /* @sic R5s150237 change 2 sic@ @sic R5s150692 change 16 sic@ @sic R5-160794, R5-162925, R5-162927, R5-162928, R5-162929: PreconditionRequired sic@ */

                                                 v_MessageBodyTx)));

    // Step 7 Optional. Receive Update, send 200 OK

    // Note: UE will send Update if Media is not yet configured

    if (not v_MediaConfigured) {

      v_SDP_LatestOffer := f_IMS_MOCallSetup_ReceiveUpdateSend200OK(v_InviteRequest, v_ContactUrl, v_SDP_Message_Step5or7, p_GeolocationInfoAvailable, p_CallType, p_PreconditionIndicationStep5and7);    /* @sic R5s201620 sic@ */

    }

    return v_SDP_LatestOffer;

  }


Change 21 – f_IMS_InviteRequest_MessageHeaderRX
	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  {

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2


    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us_Contact := *;

    var template integer v_Port_ps_Route := *;

    var template DisplayName v_DisplayName := *;

    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_Pcscf;

    var charstring v_MediaType := c_sdpApplication;

    var template (present) CallId v_CallId;

    var template (present) integer v_CSeqValue := ?;

    var template (present) From v_From;

    var template (present) To v_To;

    var template AcceptBody_List v_AcceptBodyList := {};

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    var boolean v_GRUUsInSIP := pc_IMS_GRUUsInSIP;

    var boolean v_A4 := (p_Context == A_2_1_A4);    // INVITE creating a dialog    @sic R5-166280 sic@

    var boolean v_A5 := (p_Context == A_2_1_A5);    // re-INVITE within a dialog   @sic R5-166280 sic@

    var boolean v_A6 := (p_Context == A_2_1_A6);    // emergency session in case of no registration          @sic R5-166280 sic@

    var boolean v_A19 := (p_Context == A_2_1_A19);  // emergency session in case of no registration (GIBA)   @sic R5-174545 sic@

    var boolean v_A26 := pc_IMS_Session_Timer;      // UE supports Session Timer   @sic R5-192321 sic@

    if (pc_IMS_CS_PS_SRVCCMidCall or pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_norefersub;

    }

    if (v_A4 and v_A26) {                           /* @sic R5-192321 sic@

                                                       NOTE: A4, A6, A7 and A19 are mutual exclusive; use of A20, A21 (eCall) is unclear */

      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_timer;  /* @sic R5-173024: support of session timer acc. RFC 4028  sic@

                                                                          @sic R5-180230: end of grace period sic@ */

    }

    if (v_A6 or v_A19) {  // INVITE for creating an emergency session in case of no registration

      // Route, Contact: ports to be used

      v_Port_us_Contact := ?;

      v_Port_ps_Route := tsc_IMS_PortNumber_5060;

      if (not v_A19) {                                    /* @sic R5-191118 sic@ */

        // From: Display

        v_DisplayName := cr_DisplayName("Anonymous");     /* @sic R5s140005, R5-140901, R5w140111 sic@ */

      }

      v_GRUUsInSIP := false;     /* @sic R5s160366: no GRUU in case of no registration sic@ */

    }

    else {

      // Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us_Contact := v_Protected.Port_us;

        v_Port_ps_Route := v_Protected.Port_ps;

      } else {  /* @sic R5s160366: GIBA sic@ */

        v_Port_us_Contact := ?;

        v_Port_ps_Route := tsc_IMS_PortNumber_5060 ifpresent;

      }

    }

    if (v_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := f_IMS_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX();     /* @sic R5s140235, R5s140385 change 3 MCC160 implementation sic@ */

    }

    else {                                       /* normal case: no re-invite */

      // route body acc. to conditions in A.2.1

      v_Pcscf := ?;                              /* @sic R5s140934 change 6: P-CSCF address is checked later on (f_IMS_A_2_1_Invite_CommonChecks -> f_IMS_MessageHeader_Request_CheckRoute) sic@ */

      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(v_Pcscf, v_Port_ps_Route));  /* @sic R5s140873: v_Pcscf;  R5s140934 change 6 sic@

                                                                                                  @sic R5-155174: "loose routing" parameter mandatory for any kind of call (i.e. emergency call too, in contrast to R5s150817) sic@ */

      if (not match(p_Context, (A_2_1_A6, A_2_1_A7, A_2_1_A19))) {  // => no emergency call

        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(tsc_IMS_Scscf));

      }

      if (p_A8) {

        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MediaType := tsc_MimeType_MultipartyMixed;

      }

      v_CallId := cr_CallId(?);

      v_From := cr_From(v_DisplayName, v_SipUrlFrom);

      v_To := cr_ToDef(p_ExpectedCalleeUri);

    }

    v_MessageHeader_Invite.via             := f_IMS_SipRequest_ViaRX();                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us_Contact, v_GRUUsInSIP));  /* additional checks in f_IMS_A_2_1_Invite_CommonChecks:

                                                                                                                               check of FQDN or IP address, check of feature-param for A3, A6, A10, A12

                                                                                                                               @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@

                                                                                                                               @sic R5s160366: v_GRUUsInSIP sic@ */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or v_A6 or v_A19) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (v_A6 or v_A19) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      if (not v_A5) {  /* @sic R5s140017 change 9.3 sic@ */

        v_MessageHeader_Invite.require       := cr_RequireSecAgree;

        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireSecAgree;

      }

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    }

    if (v_A4) {  /* @sic R5s140017 change 9.3 sic@ */

      v_AcceptBodyList := {cr_AcceptBody(c_sdpApplication), cr_AcceptBody(c_ims3gppApplication)};

    }

    if (pc_IMS_CS_PS_SRVCCMidCall) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.mid-call+xml"); // add another element in the array

    }

    if (pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@

      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.state-and-event-info+xml"); // add another element in the array

      v_MessageHeader_Invite.recvInfo := cr_RecvInfo_Def(tsc_IMS_InfoPackageType_3gpp);  // @sic R5-145746, R5s150704,  sic@

    }

    if (lengthof(v_AcceptBodyList) > 0) {

      if (v_A5) { // @sic R5s150971 header optional if A5 sic@

        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList)) ifpresent;

      } else {

        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList));

      }

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3 and v_A4) {                          /* @sic R5s140263 change 5, modified by R5-142949 (-> A4 only) sic@ */

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 2 sic@ */

    }

    if ((p_A3 and v_A4) or p_A11) {               /* @sic R5-142949 sic@ */

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    if (pc_EarlyMedia and not v_A5) {                                             /* @sic R5-162934: condition A16 for A.2.1  sic@

                                                                                     @sic R5-166280: "not A5" sic@ */

      v_MessageHeader_Invite.pEarlyMedia := cr_PEarlyMedia({"supported"});

    }




    return v_MessageHeader_Invite;

  }


Change 22 – f_IMS_ByeRequest_MessageHeaderRX
	  function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            IMS_CallType_Type p_CallType := NORMAL_CALL,

                                            template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit,

                                            boolean p_IsEmergencyCall := false) runs on IMS_PTC return template (present) MessageHeader

  {

    var boolean v_A1 := f_IMS_PTC_SecurityScheme_IsImsSecurity();

    var boolean v_A2 := f_IMS_PTC_SecurityScheme_IsGiba();


    var boolean v_A6 := p_IsEmergencyCall and not f_IMS_PTC_ImsInfo_IsRegistered();       /* @sic R5-172934 sic@ */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var template (present) integer v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX(dialogRemote, p_DialogIndex);  /* @sic R5s200320: f_IMS_RequestInDialog_CSeqValueRX instead of f_IMS_PTC_ImsInfo_CseqIncr sic@ */

    var template (present) RouteBody_List v_RouteBodyList;

    var template (present) Via v_Via;

    var template (present) PAccessNetworkInfo  v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    if (not fl_IMS_InviteRequestIsMO(p_InviteRequest)) {                  /* @sic R5-134116: sic@

                                                                             in case of MT call establishment the INVITE sent to the UE contains a Record-Route header with a valid route set */

      v_Via := f_IMS_SipRequest_ViaRX();                                  /* as defined in A.2.1 */

    } else {                                                              /* in case of MO call establishment the route set as for the 183 response shall be used (in reverse order) */

      v_Via := f_IMS_Via_AsInPrevInvite(v_MessageHeader_Invite.via)

    }

    v_RouteBodyList := f_IMS_RouteSet_RouteBodyList_RX(p_CallType, p_DialogIndex);              /* @sic R5s140017 MCC160 implementation: re-INVITE - Route is different for MO and MT calls sic@

                                                                                                   @sic R5s150692 change 19 sic@ */

    v_MessageHeader_Bye.via             := v_Via;

    v_MessageHeader_Bye.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Bye.toField         := f_IMS_RequestInDialog_ToHeaderRX(p_DialogIndex);     /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Bye.fromField       := f_IMS_RequestInDialog_FromHeaderRX(p_DialogIndex);   /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (v_A6) {       /* @sic R5-172934 sic@ */

      v_MessageHeader_Bye.require            := omit;

      v_MessageHeader_Bye.proxyRequire       := omit;

      v_MessageHeader_Bye.securityVerify     := omit;

    }

    else {

      if (v_A1) {

        v_MessageHeader_Bye.require          := cr_RequireSecAgree;

        v_MessageHeader_Bye.proxyRequire     := cr_ProxyRequireSecAgree;

        v_MessageHeader_Bye.securityVerify   := f_IMS_PTC_BuildSecurityVerifyHeader();          /* @sic R5-196983 sic@ */

      }

      else if (v_A2) {

        v_MessageHeader_Bye.securityVerify   := omit;

      }





    }




    if (v_A1) {

      v_MessageHeader_Bye.pAccessNetworkInfo := v_PAccessNetworkInfo;

    }

    else if (v_A2) {

      v_MessageHeader_Bye.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    return v_MessageHeader_Bye;

  }


Change 23 – f_IMS_Register_NonGIBA
	  function f_IMS_Register_NonGIBA(IMS_IpcanPortType_Type p_IpcanType := IPCAN_E,

                                  template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                  charstring p_RegisterExpiration := tsc_IMS_RegisterExpiration,

                                  boolean p_SigCompIsActive := false,

                                  boolean p_IsEmergency := false) runs on IMS_PTC return IMS_DATA_REQ

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;


    var default v_RetransmissionHandler := null;

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(p_IpcanType, p_IMS_DATA_REQ, -, -, p_SigCompIsActive, p_IsEmergency);         // check REGISTER and init GlobalInfo

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    if (fl_IMS_RetransmissionsAllowedDuringRegistration(v_IMS_DATA_REQ)) {

      v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_DATA_REQ));

    }

    f_IMS_Register_SecurityInit(v_RegisterReq);

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, p_IsEmergency);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq, p_IsEmergency, p_RegisterExpiration)))); // @sic R5s130561 sic@

    deactivate(v_RetransmissionHandler);

    return v_IMS_DATA_REQ;

  }


Change 24 – f_IMS_Registration
	  function f_IMS_Registration(IMS_IpcanPortType_Type p_IpcanType := IPCAN_E,

                              template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                              boolean p_SigCompIsActive := false,

                              charstring p_RegisterExpiration := tsc_IMS_RegisterExpiration) runs on IMS_PTC return IMS_DATA_REQ

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var boolean v_WaitForPublish := true;

    v_IMS_DATA_REQ := f_IMS_Register(p_IpcanType, p_IMS_DATA_REQ, p_SigCompIsActive, p_RegisterExpiration);

    f_IMS_Register_SubscribeNotify(v_IMS_DATA_REQ, v_WaitForPublish);

    return v_IMS_DATA_REQ;

  }


Change 25 – f_IMS_SubscribeRequest_MessageHeaderRX_C30
	  function f_IMS_SubscribeRequest_MessageHeaderRX_C30(IMS_PTC_DialogIndex_Type p_DialogIndex) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var EventType v_EventType := f_SIP_SUBSCRIBE_DialogIndex2EventType(p_DialogIndex);   /* @sic R5s200743: f_SIP_SUBSCRIBE_DialogIndex2EventType sic@ */


    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");   /* @sic R5s160045: any CSeqValue; CSeqValue to be check after the SUBSCRIBE has been effectively matched an appropriate receive statement sic@ */

    v_MessageHeader_Subscribe.fromField   := f_IMS_RequestInDialog_FromHeaderRX(p_DialogIndex);             // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.toField     := f_IMS_RequestInDialog_ToHeaderRX(p_DialogIndex);               // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.callId      := f_IMS_PTC_ImsInfo_DialogGetCallId(p_DialogIndex);              // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.expires     := cr_Expires("0");

    v_MessageHeader_Subscribe.event       := cr_EventDef(v_EventType);







    return v_MessageHeader_Subscribe;

  }


Change 26 – f_IMS_PublishRequest_MessageHeaderRX
	  function f_IMS_PublishRequest_MessageHeaderRX(template (present) SipUrl p_SipUrlTo := ?,

                                                template (present) EventType p_EventType := ?)  runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5-196983: p_SipUrlTo tocope with specific To-URI for MCX sic@ */

    var template (present) MessageHeader v_MessageHeader_Publish := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    var template (present) Expires v_Expires := cr_Expires(tsc_IMS_ExpiresDef);

    var template integer v_Port_ps := f_IMS_PTC_GetPort_ps();

    var template integer v_Port_us_Via := *;

    var template (present) SipUrl v_SipUrlFrom := ?;  /* "any IMPU within the set of IMPUs on ISIM" */




    v_MessageHeader_Publish.cSeq        := cr_CseqDef(?, "PUBLISH");

    v_MessageHeader_Publish.fromField   := cr_FromWithTag(v_SipUrlFrom);         // checked by f_IMS_PublishRequest_CheckPublicUserIdentity

    v_MessageHeader_Publish.toField     := cr_ToDef(p_SipUrlTo);                 // NOTE: absence of the To tag needs to be checked separately (f_IMS_MessageHeader_CheckTo)

    v_MessageHeader_Publish.callId      := cr_CallId(?);

    v_MessageHeader_Publish.maxForwards := cr_MaxForwardsDef;

    if (f_IMS_PTC_SecurityScheme_IsGiba()) {

      v_MessageHeader_Publish.require := omit;

      v_MessageHeader_Publish.proxyRequire := omit;

      v_MessageHeader_Publish.securityVerify := omit;

      v_MessageHeader_Publish.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

      v_MessageHeader_Publish.expires := v_Expires ifpresent;   // "same as registration timer" acc. to A.4.3

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_us_Via := v_Protected.Port_us;

      v_MessageHeader_Publish.pAccessNetworkInfo := v_PAccessNetworkInfo;

      v_MessageHeader_Publish.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Publish.expires := v_Expires ifpresent;   // "same as registration timer" acc. to A.4.3

    }

    v_MessageHeader_Publish.route   := cr_Route(cr_RouteBodyList_InitialReq(v_Port_ps));   /* P-SCCF needs to be checked after receiving (f_IMS_MessageHeader_Request_CheckRoute) */

    v_MessageHeader_Publish.via     := cr_ViaDef(-, cr_HostPort(?, v_Port_us_Via));        /* host address needs to be checked after receiving (f_IMS_MessageHeader_Request_CheckVia) */

    v_MessageHeader_Publish.event   := cr_EventDef(p_EventType);

    return v_MessageHeader_Publish;

  }


Change 27 – fl_IMS_Register_BuildAuthorizationHeader
	  function fl_IMS_Register_BuildAuthorizationHeader(RegistrationType_Type p_RegistrationType,

                                                    template (present) charstring p_Nc := "00000001") runs on IMS_PTC return template Authorization

  { /* @sic R5-160290: FBBA registration - new function sic@ */


    var charstring v_Username := f_IMS_PTC_ImsInfo_GetPrivateUserId();      /* => FBBA: px_IMS_Private_UserId (see f_IMS_PTC_Init) */

    var charstring v_Realm := f_IMS_PTC_ImsInfo_GetHomeDomainName();        /* => FBBA: px_IMS_HomeDomainName (see f_IMS_PTC_Init) */

    var charstring v_Uri := "sip:" & v_Realm;

    var template charstring v_Opaque := tsc_IMS_Opaque;

    var charstring v_Nonce;

    var charstring v_Algorithm;

    var template Authorization v_Authorization;

    select (p_RegistrationType) {

      case (initial) {




          v_Authorization := f_AuthorizationCommon(v_Username, v_Realm, v_Uri);


      }

      case (subsequent) {

        v_Nonce := f_IMS_PTC_Security_GetNonce();

        v_Algorithm := f_IMS_PTC_Security_GetHashAlgorithm();

        v_Authorization := f_AuthorizationSubsequent(v_Username, v_Realm, v_Nonce, v_Uri, v_Algorithm, p_Nc, v_Opaque);

      }

    }

    return v_Authorization;

  }


Change 28 – f_IMS_RegisterResponse_401_MessageHeaderTX
	  function f_IMS_RegisterResponse_401_MessageHeaderTX(REGISTER_Request p_RegisterRequest,

                                                      template (omit) charstring p_Nonce := omit) runs on IMS_PTC return template (value) MessageHeader

  {

    var charstring v_Nonce;

    var template (value) SecurityMechanism_List v_SecurityMech;

    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(p_RegisterRequest.msgHeader);

    if (not isvalue(p_Nonce)) {

      v_Nonce := f_IMS_PTC_Security_GetNonce();

    } else {

      v_Nonce := valueof(p_Nonce);

    }

    v_MessageHeader_Response.wwwAuthenticate := cs_WwwAuthChallenge(f_IMS_PTC_ImsInfo_GetHomeDomainName(), v_Nonce, f_IMS_PTC_Security_GetHashAlgorithm()); /* @sic R5s120727 change 2 sic@

                                                                                                                                                               @sic R5-160290 FBBA registration: f_IMS_PTC_Security_GetHashAlgorithm sic@ */


      v_SecurityMech := f_IMS_PTC_BuildSecurityServerParam();

      v_MessageHeader_Response.securityServer  := cs_SecServerDef(v_SecurityMech);


    return v_MessageHeader_Response;

  }


Change 29 – f_IMS_RegisterRequest_MessageHeaderRX_Initial
	  function f_IMS_RegisterRequest_MessageHeaderRX_Initial(InternetProtocol_Type p_Protocol,

                                                         template (present) integer p_CseqValue := ?,

                                                         template (present) SemicolonParam_List p_MandatoryContactParams := {}) runs on IMS_PTC return template (present) MessageHeader

  { 
    var boolean v_CondA3 := f_IMS_PTC_SecurityScheme_IsGiba();


    var boolean v_IsMCX := f_IMS_PTC_TestsuiteInfo_IsMCX();             /* MCX */

    var template (present) SipUrl v_SipUrlPublicUserId := ?;            /* Acc. to 24.229 cl. 5.1.1.2.1 for initial registration the UE shall populate

                                                                           'To' and 'From' with the public user identity to be registered;

                                                                           Normally this is one of the IMPUs stored on the ISIM;

                                                                           The public user identity is explicitly checked after matching the REGISTER message */

    var template (present) MessageHeader v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(v_CondA3, p_CseqValue);

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    var template GenericParam v_ViaRportParam := omit;

    if ((p_Protocol == tcp) or v_IsMCX) {       /* @sic R5-196983: rport parameter is not checked for MCX sic@ */

      v_ViaRportParam := omit;       /* @sic R5s120858 change 2; R5s130369; R5-133358: "rport" is optional for tcp (independent from A3) sic@ */

    } else {

      v_ViaRportParam := cr_GenericParam_NoValue("rport");   // @sic R5s120907 change 8 - rport has no value sic@

    }

    v_MessageHeader_Register.via         := cr_ViaDef(-, -, v_ViaRportParam);

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);                            // checked in f_IMS_Register_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);                                  // checked in f_IMS_Register_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUri_HostPort(?, *), p_MandatoryContactParams);  // additionally checked in fl_IMS_Register_ContactHeader_CheckParams, f_IMS_Register_MessageHeader_CommonChecks

    if (v_IsMCX) {                                                                 /* @sic R5-196983: pAccessNetworkInfo mandatory for MCX sic@ */

      v_MessageHeader_Register.pAccessNetworkInfo := v_PAccessNetworkInfo;

    } else {

      v_MessageHeader_Register.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;    // optional for A1, A3; cellId is not checked

    }

    if (not v_CondA3) {  // => 34.229-1 A.1.1 condition A1

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := fl_IMS_Register_BuildAuthorizationHeader(initial);  /* @sic R5-160290 FBBA registration - new function fl_IMS_Register_BuildAuthorizationHeader sic@ */

    }

    return v_MessageHeader_Register;

  }


Change 30 – f_IMS_RegisterRequest_MessageHeaderRX_Subsequent
	  function f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(integer p_NonceCount,

                                                            InternetProtocol_Type p_Protocol,

                                                            integer p_CseqValue,

                                                            template (present) SemicolonParam_List p_MandatoryContactParams := {}) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (present) MessageHeader v_MessageHeader_Register;

    var template (present) SipUrl v_SipUrlPublicUserId;

    var charstring v_NonceCountStr := "0000000" & int2str(p_NonceCount);     // assuming p_NonceCount < 9

    var IMS_ProtectedPorts_Type v_Protected;

    var boolean v_IsGiba := false;


    var template integer v_ViaPort := *;

    var template integer v_ContactPort;




      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_ContactPort := v_Protected.Port_us;

      if (p_Protocol == udp) {      // @sic R5-145727 sic@

        v_ViaPort := v_Protected.Port_us;

      }


    v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetSipUriRX(registrationId);     /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                  (in case the UE does not have an ISIM this is the temporary user id)

                                                                                  => we don't need to consider A7 (emergency registration)

                                                                                  @sic R5s160451: use of f_IMS_PTC_ImsInfo_GetSipUriRX sic@ */

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(v_IsGiba, p_CseqValue);

    v_MessageHeader_Register.via         := cr_ViaDef(-, cr_HostPort(?, v_ViaPort));                                       // @sic R5s120727 change 7; R5-145727 sic@

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUri_HostPort(?, v_ContactPort), p_MandatoryContactParams);    // additionally checked in f_IMS_Register_MessageHeader_CommonChecks; @sic R5s120727 change 7 sic@

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());                          // cell id only if applicable

    v_MessageHeader_Register.authorization      := fl_IMS_Register_BuildAuthorizationHeader(subsequent, v_NonceCountStr);  /* @sic R5-160290 FBBA registration - new function fl_IMS_Register_BuildAuthorizationHeader sic@ */


      v_MessageHeader_Register.securityVerify   := f_IMS_PTC_BuildSecurityVerifyHeader();


    return v_MessageHeader_Register;

  }


Change 31 – f_IMS_RegisterRequest_MessageHeaderRX_Derregister
	  function f_IMS_RegisterRequest_MessageHeaderRX_Derregister(integer p_CseqValue) runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5-176758: p_CseqValue, to be consistent with f_IMS_RegisterRequest_MessageHeaderRX_Subsequent sic@ */


    var boolean v_CondA3  := f_IMS_PTC_SecurityScheme_IsGiba();              /* condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA) */

    var boolean v_CondA2  := not v_CondA3;

    var template (present) MessageHeader v_MessageHeader_Register;

    var template (present) SipUrl v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetSipUriRX(registrationId);  /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                                             @sic R5-173026: acc. to IR.92 V10.0 the UE has to use the same URI as in the initial REGISTER sic@

                                                                                                             @sic R5s160451: use of f_IMS_PTC_ImsInfo_GetSipUriRX sic@ */

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    if (v_CondA2) {  // @sic R5-160290 FBBA registration sic@

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_us := v_Protected.Port_us;

    }

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(v_CondA3, p_CseqValue);

    // common parameters for A2, A3 and A15:

    v_MessageHeader_Register.fromField  := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Register.toField    := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Register.supported  := *;    // Supported: header may be missing or it may contain any value

    v_MessageHeader_Register.via        := cr_ViaDef(-, cr_HostPort(?, v_Port_us));                             // address will be checked in f_IMS_Register_MessageHeader_CommonChecks (@sic R5s120727 sic@)

    v_MessageHeader_Register.contact    := (cr_Contact_Wildcard, cr_Contact(cr_SipUri_HostPort(?, v_Port_us))); // address will be checked in f_IMS_Register_MessageHeader_CommonChecks (@sic R5s120727 sic@)

    if (not v_CondA3) {

      v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());                        // cell id only if applicable

      v_MessageHeader_Register.authorization      := fl_IMS_Register_BuildAuthorizationHeader(subsequent, ?);              /* @sic R5-160290 FBBA registration - new function fl_IMS_Register_BuildAuthorizationHeader sic@ */


        v_MessageHeader_Register.securityVerify   := f_IMS_PTC_BuildSecurityVerifyHeader();


    }

    return v_MessageHeader_Register;

  }


Change 32 – f_IMS_SubscribeRequest_MessageHeaderRX
	  function f_IMS_SubscribeRequest_MessageHeaderRX(EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  template (present) charstring p_ExpireTime := "600000",

                                                  template (present) SemicolonParam_List p_MandatoryContactParams := {})

    runs on IMS_PTC return template (present) MessageHeader

  {

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    var template integer v_Port_ps := *;

    var template integer v_Port_us_Via := *;

    var template integer v_Port_us_Contact := ?;

    var boolean v_CondA2 := f_IMS_PTC_SecurityScheme_IsGiba();       // GIBA


    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (v_CondA2 ) {   // GIBA or FBBA   @sic R5-160290 FBBA registration sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      if (v_CondA2) {

        v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

      } else {

        v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo;

      }

    }

    else {                        // IMS security

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_ps := v_Protected.Port_ps;

      v_Port_us_Via := v_Protected.Port_us;

      v_Port_us_Contact := v_Protected.Port_us;

      v_MessageHeader_Subscribe.require := cr_RequireSecAgree;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireSecAgree;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo;

    }

    v_MessageHeader_Subscribe.route   := cr_Route(cr_RouteBodyList_InitialReq(v_Port_ps));                  /* @sic R5s140934 change 1 sic@ P-CSCF address is checked by f_IMS_MessageHeader_Request_CheckRoute */

    v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Port_us_Via));                       /* host address needs to be checked after receiving (f_IMS_MessageHeader_Request_CheckVia) @sic R5s120727 change 6.1 sic@ */

    v_MessageHeader_Subscribe.contact := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us_Contact), p_MandatoryContactParams);     /* host address needs to be checked after receiving (f_IMS_MessageHeader_CheckContactAddr)

                                                                                                                                         @sic R5s120727 change 6.1 sic@

                                                                                                                                         @sic R5s130925 change 1: UE may use any port as server port sic@

                                                                                                                                         @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(c_xmlreginfoApplication) ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);      // !!!! should be removed !!!!

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary");

      }

      case ("presence") {           // SUBSCRIBE for presence event package MCX TS 36.579-1 Table 5.5.2.14-1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(tsc_MimeType_Pidf);

      }

      case ("xcap-diff") {          // SUBSCRIBE for xcap-diff event package, MCX TS 24.379 clause 6.3.5

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(tsc_MimeType_XcapDiff);

      }

      case ("poc-settings") {       // SUBSCRIBE for poc-settings event package, MCX TS 24.379 clause 7.3.6

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(tsc_MimeType_pocSettings);

      }                                              

    }

    return v_MessageHeader_Subscribe;

  }


Change 33 – f_IMS_ReferRequest_MessageHeaderRX
	  function f_IMS_ReferRequest_MessageHeaderRX(template (present) SipUrl p_RequestUri,

                                              template (present) SipUrl p_ReferToUri,

                                              boolean p_ReferCreatingDialog,

                                              template (present) integer p_CSeqValue := ?,

                                              IMS_CallType_Type p_CallType := CONFERENCE_CALL) runs on IMS_PTC return template (present) MessageHeader

  {

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Refer := cr_MessageHeader_Dummy;

    var template (present) RouteBody_List v_RouteBodyList;

    var template (present) SipUrl v_SipUrlFrom := ?;

    var template (present) From v_From;

    var template (present) To v_To;

    var template (present) CallId v_CallId;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := ?;   // Acc. A.2.10 mandatory even for GIBA

    var template integer v_Port_ps := *;




    if (not v_IsGiba) {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_us := v_Protected.Port_us;

      v_Port_ps := v_Protected.Port_ps;

    }

    if (p_ReferCreatingDialog) {

      // FFS: currently not used (p_ReferCreatingDialog is always false)

      v_RouteBodyList := {                    // same as in INVITE

        cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps)),

        cr_RouteBody(*, cr_SipUri_HostPort_lr(tsc_IMS_Scscf))

      };

      v_From := cr_From(*, v_SipUrlFrom);

      v_To := cr_ToDef(p_RequestUri);

      v_CallId := cr_CallId(?);

    } else {

      v_RouteBodyList := f_IMS_RouteSet_RouteBodyList_RX(p_CallType);       /* @sic R5s160011: p_CallType sic@ */

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();                      /* @sic R5-151954 sic@ */

    }

    v_MessageHeader_Refer.via          := f_IMS_SipRequest_ViaRX(?);  // same as for INVITE; but p_Port_us is mandatory

    v_MessageHeader_Refer.route        := cr_Route(v_RouteBodyList);

    v_MessageHeader_Refer.toField      := v_To;

    v_MessageHeader_Refer.fromField    := v_From;

    v_MessageHeader_Refer.callId       := v_CallId;

    v_MessageHeader_Refer.cSeq         := cr_CseqDef(p_CSeqValue, "REFER");

    v_MessageHeader_Refer.maxForwards  := cr_MaxForwardsDef;

    v_MessageHeader_Refer.contact      := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us));     /* additional checks required after message has been received

                                                                                                        @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

    v_MessageHeader_Refer.referTo      := cr_ReferTo(p_ReferToUri);

    if (not v_IsGiba) {

      v_MessageHeader_Refer.require         := cr_RequireSecAgree;

      v_MessageHeader_Refer.proxyRequire    := cr_ProxyRequireSecAgree;

      v_MessageHeader_Refer.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Refer.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    } else {

      v_MessageHeader_Refer.securityVerify  := omit;

    }

    return v_MessageHeader_Refer;

  }


Change 34 – fl_ConferenceEventPackage_SubscribeRequest_MessageHeaderRX
	  function fl_ConferenceEventPackage_SubscribeRequest_MessageHeaderRX(template (present) SipUrl p_RequestUri)

    runs on IMS_PTC return template (present) MessageHeader

  {

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    var template (present) SipUrl v_SipUrlFrom := ?;

    var template (present) RouteBody_List v_RouteBodyList;

    var template (present) From v_From;

    var template (present) To v_To;

    var template (present) CallId v_CallId;

    var template (present) integer v_Port_us := ?;

    var template integer v_Port_ps := *;

    var IMS_ProtectedPorts_Type v_Protected;




    if (v_IsGiba) {

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_us := v_Protected.Port_us;

      v_Port_ps := v_Protected.Port_ps;

      v_MessageHeader_Subscribe.require := cr_RequireSecAgree;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireSecAgree;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    v_RouteBodyList := {                    // same as in REFER

      cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps)),

      cr_RouteBody(*, cr_SipUri_HostPort_lr(tsc_IMS_Scscf))

    };

    v_From := cr_From(*, v_SipUrlFrom);

    v_To := cr_ToDef(p_RequestUri);

    v_CallId := cr_CallId(?);

    v_MessageHeader_Subscribe.via         := f_IMS_SipRequest_ViaRX(?);     /* same as for INVITE; but p_Port_us is mandatory

                                                                               @sic R5-153986: common review of implementation sic@ */

    v_MessageHeader_Subscribe.route       := cr_Route(v_RouteBodyList);

    v_MessageHeader_Subscribe.toField     := v_To;

    v_MessageHeader_Subscribe.fromField   := v_From;

    v_MessageHeader_Subscribe.callId      := v_CallId;

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.expires     := cr_Expires(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef("conference");

    v_MessageHeader_Subscribe.accept      := cr_AcceptDef("application/conference-info+xml");

    v_MessageHeader_Subscribe.contact := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us));     /* host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

                                                                                                       @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

    return v_MessageHeader_Subscribe;

  }


Change 35 – f_IMS_PrackRequest_MessageHeaderRX
	  function f_IMS_PrackRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                              template (value) MessageHeader p_MessageHeader_PrevReliableResponse,

                                              template ContentType p_ContentType := *
) runs on IMS_PTC return template (present) MessageHeader

  {

    var MessageHeader v_MessageHeader_PrevReliableResponse := valueof(p_MessageHeader_PrevReliableResponse);

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Prack := cr_MessageHeader_Dummy;

    var CSeq v_CSeq := v_MessageHeader_PrevReliableResponse.cSeq;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);                   /* @sic R5s130497 change 3; R5-133188 sic@ */

    var integer v_ResponseNumber := v_MessageHeader_PrevReliableResponse.rSeq.responseNum;   /* @sic R5s130333 change 5.2 sic@ */

    v_MessageHeader_Prack.via             := f_IMS_Via_AsInPrevInvite(v_MessageHeader_Invite.via); // @sic R5s120530 change 3.9 sic@

    v_MessageHeader_Prack.route           := fl_IMS_Route_ReverseOrderOfRecordRouteRX(v_MessageHeader_PrevReliableResponse.recordRoute);

    v_MessageHeader_Prack.toField         := f_IMS_RequestInDialog_ToHeaderRX();     /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Prack.fromField       := f_IMS_RequestInDialog_FromHeaderRX();   /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Prack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Prack.cSeq            := cr_CseqDef(v_CSeqValue, "PRACK");

    v_MessageHeader_Prack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Prack.rAck            := cr_RAckDef(v_ResponseNumber, v_CSeq.seqNumber, v_CSeq.method);

    v_MessageHeader_Prack.contentType     := p_ContentType;

    if (not f_IMS_PTC_SecurityScheme_IsGiba()) {  /* @sic R5s170099 sic@ */

      v_MessageHeader_Prack.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    } else {

      v_MessageHeader_Prack.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType()) ifpresent;

    }




    return v_MessageHeader_Prack;

  }


Change 36 – f_IMS_REGISTER_InitialRequest
	  function f_IMS_REGISTER_InitialRequest(IMS_IpcanPortType_Type p_IpcanType := IPCAN_E,

                                         template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                         template (present) SemicolonParam_List p_MandatoryContactParams := {},

                                         template (omit) charstring p_ExplicitExpiry := omit,

                                         boolean p_SigCompIsActive := false,

                                         boolean p_CondA7 := false,

                                         boolean p_CondA3 := false
) runs on IMS_PTC return IMS_DATA_REQ

  { 

    var IPCAN_INFO_Type v_IpcanInfo;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_RoutingInfo_Type v_RoutingInfo_UL;

    var InternetProtocol_Type v_Protocol;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var MessageHeader v_MessageHeader;

    var template (present) integer v_CseqValue := ?;

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ, v_CseqValue);   /* @sic R5-176758: v_CseqValue sic@ */

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    select (p_IpcanType) {

      case (IPCAN_E) {

        v_IpcanInfo := f_IMS_IPCAN_Query(IPCAN);

      }

      case else {

        v_IpcanInfo := f_IMS_IPCAN_Query(OtherIPCAN);

      }

    }

    f_IMS_PTC_SetIpcanInfo(v_IpcanInfo);

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Initial(v_Protocol, v_CseqValue, p_MandatoryContactParams));         /* @sic R5s130369 sic@

                                                                                                                                                    @sic R5-176758: v_CseqValue sic@

                                                                                                                                                    @sic R5s200000: p_MandatoryContactParams sic@*/

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    v_RoutingInfo_UL := v_IMS_DATA_REQ.RoutingInfo;

    v_UE_Address := v_RoutingInfo_UL.UE_Address;

    v_NW_Address := v_RoutingInfo_UL.NW_Address;

    v_Protocol := v_RoutingInfo_UL.Protocol;

    f_IMS_PTC_ImsInfo_Init(v_UE_Address, v_NW_Address, v_IMS_DATA_REQ.Request.Register, p_SigCompIsActive);

    // additional checks

    f_IMS_Register_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry);

    f_IMS_Register_MessageHeader_CheckPublicUserIdentity(v_MessageHeader, p_CondA7, p_CondA3);  // check 'To' and 'From' field (needed for initial registration only); @sic R5s140356: p_CondA3 sic@

    if (p_SigCompIsActive) {

      f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

    }

    return v_IMS_DATA_REQ;

  }


Change 37 – fl_IMS_RegisterRequest_MessageHeaderRX_Base
	  function fl_IMS_RegisterRequest_MessageHeaderRX_Base(boolean p_CondA3 := false,

                                                       template (present) integer p_CseqValue
) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (present) MessageHeader v_MessageHeader_Register := cr_MessageHeader_Dummy;

    var OptionTag_List v_SupportedOptions := { "path" };                /* @sic R5s150721 - Additional change: "path" shall always be included sic@ */

    var boolean v_IsMCX := f_IMS_PTC_TestsuiteInfo_IsMCX();             /* MCX */

    if (v_IsMCX) {                                                      /* @sic R5-196983: MCX sic@ */

      v_SupportedOptions[1] := "timer";

    } else if (pc_IMS_GRUUsInSIP) {

      v_SupportedOptions[1] := "gruu";

    }

    v_MessageHeader_Register.route       := omit; // @sic R5-134659 sic@

    v_MessageHeader_Register.cSeq        := cr_CseqDef(p_CseqValue, "REGISTER");

    v_MessageHeader_Register.expires     := *;                                  // as the 'expires' information can be included in 'expires' as well as in 'contact' (or in both) it needs to be checked seperately

    v_MessageHeader_Register.supported   := cr_Supported(v_SupportedOptions);   // @sic R5-196983: cr_Supported instead of cr_Supported_Register sic@

    v_MessageHeader_Register.callId      := cr_CallId(?);                       // per default call id itself is not checked at all

    v_MessageHeader_Register.maxForwards := cr_MaxForwardsDef;

    if (p_CondA3 ) {                                // GIBA or FBBA

      v_MessageHeader_Register.require := omit;

      v_MessageHeader_Register.proxyRequire := omit;

      v_MessageHeader_Register.securityClient := omit;

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := omit;              /* A14/A15: gets overwritten with expected header */

    }

    else {

      v_MessageHeader_Register.require := cr_RequireSecAgree;

      v_MessageHeader_Register.proxyRequire := cr_ProxyRequireSecAgree;

      v_MessageHeader_Register.securityClient := cr_SecClientMandatoryFields;   // further checks in fl_IMS_SecurityClient_Check

    }

    return v_MessageHeader_Register;

  }


Change 38 – f_IMS_A_2_1_Invite_CommonChecks
	  function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                           boolean p_A3 := false,

                                           boolean p_A8 := false,

                                           boolean p_A10 := pc_IMS_Video_FeatureTag,

                                           boolean p_A11 := false,

                                           boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                           boolean p_OriginatingIdentificationRestriction := false,


                                           boolean p_A18 := pc_BSRVCC) runs on IMS_PTC

  { 
    var MessageHeader v_MessageHeader := p_IMS_INVITE_REQ.Request.Invite.msgHeader;

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_AcceptContactParams := {};

    var InternetProtocol_Type v_Protocol := p_IMS_INVITE_REQ.RoutingInfo.Protocol;

    var SipUrl v_SipUrlFrom := f_Addr_Union_GetSipUrl(v_MessageHeader.fromField.addressField);

    var SipUrl v_SipUrlPreferred;

    var template SipUrl v_ExpectedEmergencyUrl := f_IMS_PTC_ImsInfo_GetSipUriRX(emergencyId, *);

    var PPreferredIDValue_List v_PreferredIDValueList;

    var CharStringList_Type v_PreferredPublicUserIdList;

    var boolean v_EmergencyCallWithoutRegistration := (p_Context == A_2_1_A6);

    var boolean v_AllowFQDN := not v_EmergencyCallWithoutRegistration;  // for A6 there is no FQDN for the UE

    var boolean v_CheckRport := v_EmergencyCallWithoutRegistration;

    var boolean v_A12 := p_SrvccAlertingSupported and (p_Context != A_2_1_A5);  /* @sic R5s140370 change 7 sic@ */

    var integer i;

    var template (present) GenValue v_SipInstanceValue;

    var template integer v_Port_us_Via := *;

    if ((v_EmergencyCallWithoutRegistration and (v_Protocol==udp))) {  /* @sic R5-172412 sic@ */

      v_Port_us_Via := tsc_IMS_PortNumber_5060;

    }

    if (ispresent(v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams)) {

      v_ContactParams := v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams;

    }

    if (ispresent(v_MessageHeader.acceptContact.acValues[0].acRcParams)) {

      v_AcceptContactParams := v_MessageHeader.acceptContact.acValues[0].acRcParams;

    }

    // ****  Common checks ****

    f_IMS_MessageHeader_Request_CheckVia(v_MessageHeader, v_Protocol, v_AllowFQDN, v_CheckRport, v_Port_us_Via);   /* check whether via contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

                                                                                                                      @sic R5-172412: v_Port_us_Via sic@ */

    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader, v_EmergencyCallWithoutRegistration);      /* check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

                                                                                                       @sic R5s160366: v_EmergencyCallWithoutRegistration sic@ */

    if (p_Context != A_2_1_A5) {

      f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);                                        /* check that there is no To tag */

      f_IMS_MessageHeader_Request_CheckRoute(v_MessageHeader);                                      /* @sic R5s140934 change 7 sic@ */

    }

    // ****  Check URIs in From header, P-Preferred-Identity header and further headers ****

    select (p_Context) {

      case (A_2_1_A4) {

        // Check Call-Id

        if (match(v_MessageHeader.callId.callid, f_IMS_PTC_ImsInfo_GetCallIdRegistration())) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'CallId' in INVITE (has to be different than in previous REGISTER)");

        }

        // Additional checks regarding SIP URI in the From header @sic R5-130750 change 3 sic@

        if (not p_OriginatingIdentificationRestriction) { /* @sic R5-160975, R5s160687: p_OriginatingIdentificationRestriction sic@ */

          if (not f_IMS_PublicUserId_Match(f_IMS_PTC_ImsInfo_GetAllIMPUs(), v_SipUrlFrom)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid 'From'");

          }

          if (ispresent(v_MessageHeader.pPreferredID.pPreferredIDValueList)) {

            v_PreferredIDValueList := v_MessageHeader.pPreferredID.pPreferredIDValueList;

            for (i:=0; i < lengthof(v_PreferredIDValueList); i := i+1) {

              v_PreferredPublicUserIdList[i] := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_PreferredIDValueList[i]));

            }

            if (not f_IMS_PublicUserId_Match(v_PreferredPublicUserIdList, v_SipUrlFrom)) {

              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'From': does not match with P-Preferred-Identity");

            }

          }

        }

      }

      case (A_2_1_A5) {

        /* @sic R5s140017 MCC160 implementation: no further checks for A5 are needed in this function anymore sic@ */

      }

      case (A_2_1_A6) {

        v_SipInstanceValue := cr_GenValueQuoted(f_SIP_InstanceId()); /* @sic R5s130756 change 11; R5w140111: GenValue; R5s141212  sic@ */

        if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("+sip.instance", v_SipInstanceValue))) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "IMEI code of the UE, coded as specified in draft-montemurro-gsma-imei-urn, not matching");

        }

      }

      case (A_2_1_A7) {

        if (not match(v_SipUrlFrom, v_ExpectedEmergencyUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'From'");

        }

        v_SipUrlPreferred := f_Addr_Union_GetSipUrl(v_MessageHeader.pPreferredID.pPreferredIDValueList[0]);

        if (not match(v_SipUrlPreferred, v_ExpectedEmergencyUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'P-Preferred-Identity'");

        }

      }

      case (A_2_1_A19)  { // @sic R5-174545, Change 2 R5s180463 sic@

        // Via, Route and Contact header are checked in function f_IMS_InviteRequest_MessageHeaderRX

        if (not match(v_SipUrlFrom, v_ExpectedEmergencyUrl)) {          // @sic R5-191118 sic@

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A19: invalid 'From'");

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "unknown context");

      }

    }

    // ****  Check further Contact feature-param acc. to further conditions of A.2.1:  ****

    //

    if (p_A3) {                         // A.1.2 condition A3: MTSI

      if (not match(v_ContactParams, cr_FeatureParamList_3gpp_icsi_ref)) {            // @sic R5s130738: string needs to be quoted; R5-142247 sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param in Contact header");

      }

      if (p_Context == A_2_1_A4) {      // @sic R5-142949 sic@

        if (not match(v_AcceptContactParams, cr_FeatureParamList_3gpp_icsi_ref)) {    // @sic R5s130738: string needs to be quoted; R5-142247 sic@

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param in AcceptContact header");

        }

      }

    }

    if (p_A8) {                         // A.1.2 condition A8: check geolocation and geolocationRouting @sic R5-133151 sic@

      /* NOTE: v_MessageHeader.geolocation and geolocationRouting are checked already by template returned by f_IMS_InviteRequest_MessageHeaderRX:

       *       geolocation and geolocationRouting to be present. @sic R5-133151 sic@ */

      if (not match(v_MessageHeader.geolocationRouting.state, GEOLOCATION_ROUTING_YES_E)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A8: Wrong geolocation-routing");

      }

    }

    if (p_A10) {             /* A.1.2 condition A10: UE supports video feature tag

                                A.1.2 condition A11: INVITE for creating a video call

                                @sic R5-153766: "p_A10 and p_A11 and (p_Context != A_2_1_A6) and (p_Context != A_2_1_A7)" sic@

                                @sic R5-160553: "p_A10 and p_A11" sic@

                                @sic R5-176371: "p_A10" for ContactParams, "p_A10 and p_A11" for AcceptContactParams sic@ */

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10, A11: Missing feature-param in Contact header");

      }

      if (p_A11 and not match(v_AcceptContactParams, cr_VideoInContactParams)) {  /* @sic R5-176371 sic@ */

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10, A11: Missing feature-param in Accept-Contact header");

      }

    }

    if (f_IMS_PTC_AudioFeatureTagIsRequired() and not match(v_ContactParams, cr_AudioInContactParams)) {                  // check 'feature-param'  @sic R5-165943, R5-170564 sic@ @sic R5-173021 sic@  @sic R5-177127 sic@

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "missing audio feature-param in Contact header");

    }

    if (v_A12) {                                                // A.1.2 condition A12: INVITE for creating a voice or video call and UE supports g.3gpp.srvcc-alerting media feature tag

      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParamNoValue(tsc_IMS_FeatureCap_SrvccAlerting))) { /* @sic R5s140370 change 7: tsc_IMS_FeatureCap_SrvccAlerting sic@ */

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A12: Missing feature-param");

      }

    }









    if (p_A18 and (p_Context != A_2_1_A5)) {                    // A.1.2 condition A18: UE indicates g.3gpp.ps2cs-srvcc-orig-pre-alerting media feature tag in INVITE request @sic R5-171486 sic@ @sic R5-182793 sic@

      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParamNoValue(tsc_IMS_FeatureCap_SrvccPreAlerting))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A18: Missing feature-param");

      }

    }

  }


Change 39 – fl_PublicUserIdentity_Check
	  function fl_PublicUserIdentity_Check(SipUrl p_SipUrl,

                                       boolean p_CondA7 := false,

                                       boolean p_CondA3 := false,


                                       boolean p_EmergencyRegistrationWithoutPriorRegularRegistration := false) runs on IMS_PTC return boolean

  {

    var boolean v_Result := true;

    var IMS_UriInfo_UriType_Type v_UriType;





      if (p_CondA7) {                /* emergency registration */

        if (f_IMS_PTC_ImsInfo_GetUEwithISIM()) {  /* UE has ISIM @sic R5-203077: f_IMS_PTC_ImsInfo_GetUEwithISIM instead of px_UEwithISIM sic@ */

          v_UriType := emergencyId;               /* A.1.1 Note 4: UE is using ISIM */

        }

        else {                                    /* UE has no ISIM */

          if (not p_EmergencyRegistrationWithoutPriorRegularRegistration) {    /* @sic R5-176207 sic@ */

            v_UriType := defaultId;               /* A.1.1 Note 4 UE has no ISIM but successfully performed IMS registration with the IM CN subsystem before */

          } else {

            v_UriType := temporaryId;

          }

        }

        v_Result := f_SIP_MatchURI(f_IMS_PTC_ImsInfo_GetUriString(v_UriType), p_SipUrl);

      }

      else {                         /* no emergency registration */

        if (not p_CondA3 and f_IMS_PTC_ImsInfo_GetUEwithISIM()) {                            /* A1 and UE has ISIM  @sic R5-203077: f_IMS_PTC_ImsInfo_GetUEwithISIM instead of px_UEwithISIM sic@ */

          v_Result := f_IMS_PublicUserId_Match(f_IMS_PTC_ImsInfo_GetAllIMPUs(), p_SipUrl);   /* any IMPU within the set of IMPUs on ISIM

                                                                                                NOTE: for MCX there is only one IMPU (px_MCPTT_Client_A_ID, see f_IMS_PTC_Init) as according to 36.579-1 Table 5.5.2.13-1 */

        }

        else {                                   /* A3 or no ISIM */

          v_Result := f_SIP_MatchURI(f_IMS_PTC_ImsInfo_GetUriString(temporaryId), p_SipUrl); /* public user identity derived from IMSI */

        }

      }


    return v_Result;

  }


Change 40 – f_IMS_Register_MessageHeader_CheckPublicUserIdentity
	  function f_IMS_Register_MessageHeader_CheckPublicUserIdentity(MessageHeader p_MessageHeader,

                                                                boolean p_CondA7 := false,

                                                                boolean p_CondA3 := false,

                                                                boolean p_CondA14 := false) runs on IMS_PTC

  {

    var SipUrl v_SipUrlFrom := f_Addr_Union_GetSipUrl(p_MessageHeader.fromField.addressField);  /* @sic R5s130235 additional changes sic@ */

    var SipUrl v_SipUrlTo := f_Addr_Union_GetSipUrl(p_MessageHeader.toField.addressField);      /* @sic R5s130235 additional changes sic@ */

    if (not fl_PublicUserIdentity_Check(v_SipUrlTo, p_CondA7, p_CondA3, p_CondA14)) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid public user identity in 'toField' of the REGISTER request");

    }

    if (f_SIP_SipUrl2Str(v_SipUrlFrom) != f_SIP_SipUrl2Str(v_SipUrlTo)) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "public user identity is different in From and To header of the REGISTER request");

    }

  }


Change 41 – f_MTC_MainLoop_IMS
	  function f_MTC_MainLoop_IMS(timer p_GuardTimer) runs on MTC_IMS

  {

    var boolean v_IMS1_Done := (vc_Components.IMS1 == null);    // @sic R5s130011 sic@

    var boolean v_IMS2_Done := (vc_Components.IMS2 == null);    // @sic R5s130011 sic@

    f_MTC_BASE_Init();

    while (not (v_IMS1_Done and v_IMS2_Done)) {       // @sic R5s130011 sic@

      alt {

        [not v_IMS1_Done] vc_Components.IMS1.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS1_Done := true;

        }

        [not v_IMS2_Done] vc_Components.IMS2.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS2_Done := true;

        }

        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_EUTRA);    // handle upper tester for PTC 0 (EUTRA)

        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_UTRAN);    // handle upper tester for PTC 1 (UTRAN)

        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_IMS1);     // handle upper tester for PTC 4 (IMS1)

        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_IMS2);     // handle upper tester for PTC 5 (IMS2)

        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_WLAN);     // handle upper tester for PTC 6 (WLAN @sic R5w160212 sic@)


        [] a_MTC_ReceiveFromPTC_SendToUT(tsc_MTC_PortIndex_NR);       // handle upper tester for PTC 8 (NR5GC @sic R5s201327 change 6 sic@)

        [] a_MTC_ReceiveFromUT_SendToPTC();

        [] p_GuardTimer.timeout {f_MTC_Stop(fail);}    /* acc. to the assumption that preamble and postamble are short compared to the testbody

                                                          a FAIL verdict is assigned (rather than an INCONC);

                                                          @sic R5s140243 - 2nd timeout removed sic@ */

        [] any component.killed {

          f_MTC_Stop(inconc);

        }

      }

    }

    f_MTC_Stop(pass);   /* Note: "pass" overrules "none" only */

  }


Change 42 – module MTC_Base
	  const integer tsc_NoOfCordPorts := 8;

	

	  const integer tsc_MTC_PortIndex_NR := 7;


Change 43 – fl_IMS_GetDrbRoutingInfo
	  function fl_IMS_GetDrbRoutingInfo() runs on IMS_PTC return template (value) IP_DrbInfo_Type

  {

    var template (omit)  IPCAN_RAN_Type v_RanType := f_IMS_PTC_GetRanType();   /* @sic R5-163201: IPCAN_RAN_Type instead of charstring sic@ */

    var IP_DrbId_Type v_DrbId;

    var template (value) IP_DrbInfo_Type v_DrbRoutingInfo;

    var WLAN_CellId_Type v_CellId;

    if (isvalue(v_RanType)) {

      select(valueof(v_RanType)) {

        case (EUTRA_FDD, EUTRA_TDD) {

          v_DrbId := f_EUTRA_NB_EpsBearerAssociatedDRB(tsc_EpsDefaultBearerId2ndPDN);    /* DRB Id for 2nd PDN; @sic R5s141351 change 4 sic@ */

          v_DrbRoutingInfo := cs_DrbInfo_EUTRA(eutra_Cell1, v_DrbId);

        }

        case (WLAN) {

          v_CellId := wlan_Cell27;

          v_DrbRoutingInfo:= cs_DrbInfo_WLAN(v_CellId);

        }




        case else {

          FatalError(__FILE__, __LINE__, "RAN type not supported by current implementation");

        }

      }

    } else {

      FatalError(__FILE__, __LINE__, "Unknown RAN type");

    }

    return v_DrbRoutingInfo;

  }


